
ABSTRACT
It is sometimes the case there is no general consensus on the best way to treat patients suffering from an illness or 
disorder.  Consider a scenario where there are two competing treatments and it has been shown that one treatment 
works for some patients while another treatment works well for others.   In such cases we may want to define an 
algorithm (or treatment regime) that dictates treatment based on individual characteristics.  The optimal treatment 
regime would be the so-call “best” algorithm and minimizes the number of poor outcomes.  It can be very difficult to 
assess the overall public health impact of such an algorithm.

Attributable benefit (AB) of a treatment regime is a useful metric for assessing such algorithms by looking at the 
proportion of poor outcomes that could have been prevented had the algorithm or regime of interest been 
implemented.   Here we will give an overview of the assumptions for using the attributable benefit measure and 
discuss two SAS macros for estimating AB for the optimal treatment regime.  These macros are designed for the 
scenario where there is binary treatment, binary outcome, and possibly many different covariates.  The first macro 
uses an estimator based on a logistic regression model on the outcome of interest while the second macro augments 
this estimator with a propensity score model to provide doubly robust protection from model misspecification.

INTRODUCTION
In many medical research scenarios, observational data is used to determine the effectiveness of different treatment 
strategies on patient health.  In some cases, a treatment that works well for some patients may not work as well for 
others.  When treatments compete like this, it may be that the best way to deal with individual patients is to develop a 
strategy or policy that assigns treatment to the individual based on his or her risk factors.  The hope is that overall 
health will be improved if each patient receives the treatment that is right for them.

While the terminology varies across disciplines (treatment strategy, policy, algorithm, and regime are just a few 
terms), the overall goal of this type of research is to identify which treatment (or combination of treatments) has the 
most impact on overall public health.  Thus the mission here is two-fold:  first to identify the best strategy for dealing 
with individual patients and second to assess the public health impact of such a strategy.  The focus of this paper is 
later, to measure and quantify the public health impact of treatment policies which may or may not treat to the 
individual patient.

Define a treatment regime as an algorithm or set of rules for treating patients.  In addition, we will also call the “best” 
algorithm for dealing with patients to be the optimal treatment regime.  We will assume that the optimal treatment 
regime is “best” because it minimizes the number of poor outcomes.  Generally speaking, there is no limit on the 
possible number of different treatment regimes to handle patients who have some disease or disorder.  So instead of 
clinical testing, observational databases have become a key component for such research, helping to identify 
potential subgroups of individuals who seem to respond differently to certain treatments than others.  

This paper will discuss one method for estimating the optimal treatment regime and two ways of estimating the public 
health impact.  The current discussion is restricted to the scenario where the outcome is binary (i.e. disease/no 
disease or death/no death), treatment is binary, and there are any number of patient covariates or factors.  SAS 
Macros have been developed to perform these analyses given certain assumptions about the data are met.

CAUSAL INFERENCE
The underlying theory behind estimating the effects of interest is grounded in causal inference, more specifically 
counter-factual data analysis.  As such it may be important to review some of the necessary background information. 
Suppose we have data of the form {Y,T,X} where Y is a 0/1 binary outcome variable (Y = 1 is an indicator of poor 
outcome).  T is a 0/1 binary treatment variable (0 is the standard treatment and 1 is the alternate treatment), and X is 
one or more confounding variables.  Let’s say that the primary interest is in determining the causal effect of treatment 
on the chance of a poor outcome, going well beyond the idea that treatment and outcome are possibly associated. 
This is certainly much easier in the clinical setting where a proper experiment can be performed, but the need to 
make such conclusions from observational data has become increasingly more important.  So consider two 
hypothetical worlds, one where all patients receive the standard treatment and another where all patients receive the 

alternate treatment.  In each hypothetical world there would be an outcome associated for each treatment, so instead 
of considering Y itself consider instead a set of potential outcomes of the form

.

Which denotes the set of outcomes that would have occurred in each hypothetical world; these quantities are also 
referred to as counterfactuals given that in practice only one potential outcome can actually be observed while the 
other is counter to what actually happened.  Using this idea of potential outcomes, Rubin (1974) was able to form 
causal relationships between T and Y given two broad assumptions:

1. The observed outcome for an individual is the same as the potential outcome that corresponds to the 
treatment the individual actually received (also known as the Stable Unit Treatment Value Assumption).

2. Treatment assignment to an individual is independent of the potential outcomes W given the covariate 
information in X.  This assumption is also known as no unmeasured confounding because it is only possible 
if all the information on deciding treatment is captured by the covariates in X.

From this Rubin was able to show that probabilities of potential outcomes can be written as the expectation of 
observed quantities.  For example, we can write

.

In this way we can now extend the discussion beyond the naïve scenario of simply giving treatment 0 or 1 to all 
patients and begin to talk about policies where different patients.  

OPTIMAL TREATMENT REGIME
A dynamic treatment regime (or adaptive strategy) is formally defined as a set of decision rules for how treatments 
are to be assigned to an individual based on factors that vary between individuals.  The term is traditionally used to 
express strategies that are time dependent, that is to say that a dynamic treatment regime is usually a policy set over 
multiple time points.  But the term can also be applied to so-called point-exposure studies such as the type discussed 
here.  The interested reader can see Moodie et al. (2007) for more information.  Currently there is a lot of research 
delving into the best way to find the optimal treatment regime and there may certainly be better methods for 
estimating the optimal treatment regime than using an outcome regression model.  The goal of this research is in 
finding ways to estimate that policy’s public health impact, and as such the underlying theory is still appropriate even 
if there is an alternate method for determining which strategy is optimal.

As stated earlier, we will use logistic regression models to estimate the optimal treatment regime, so suppose that we 
posit that our data comes from the following outcome regression model:

.

Estimates of this model can be obtained from data easily using Proc Logistic in SAS.  Suppose that we define those 
estimates as 

.

Let g(X) be a function that assigns treatment based on the covariates in X.  Define the optimal treatment regime as 
the one that minimizes the overall probability of a poor outcome. Now let’s denote the optimal treatment regime as 

We generally abbreviate this as gopt , note the optimal treatment regime is one that defaults to the standard treatment 
and gives the alternate treatment in cases where the estimated probability of a poor outcome is lower than using the 
standard treatment.  It is important to note that if the optimal treatment is to give all patients the standard (or 
alternate) treatment then gopt is the same as a static regime where everyone receives the dominate treatment, gdom.  

In many real world cases, the optimal treatment strategy is in fact to give all patients the dominant treatment.  In such 
cases where there is a subgroup or cohort of patients that should receive the non-dominant treatment, then under this 
framework they would be detected via a significant covariate by treatment interaction.  The use of treatment by 
covariate interactions can be somewhat controversial and the best course of action may be to only proceed in such 

cases where there is a true demonstrated effect that makes sense from both the analytic and clinical perspectives. 
The reader should note that in the example listed below, the true utility of some estimates of attributable benefit are 
only found in those cases where gdom ≠ gopt.  

ATTRIBUTABLE BENEFIT
Define the attributable benefit of a treatment regime as the proportion of poor outcomes that could have been 
prevented had the treatment strategy g(X) been implemented.  More formally we write:

where

.

We will denote ABdom as the attributable benefit of gdom and ABopt as the attributable benefit of gopt.  

REGRESSION BASED ESTIMATE
Using the logistic model estimates from above one estimate of ABopt is 

where we define the function h as

The function h is an estimate of the proportion of poor outcomes that would have occurred had the optimal treatment 
regime been implemented.  The simple sample proportion of poor outcomes has been chosen for our estimate of 
P(Y=1).  

Brinkley et al. (2010) showed that a log transformed version of 1 minus the above estimate (log of the h minus the 
sample mean of y) is asymptotically normal and calculates a standard error using the theory of influence functions 
and the sandwich estimator of variance.  The interested reader may see Tsiatis (2006) for a complete overview of the 
theory of influence functions and the methods used in both SAS macros discussed here. Two sets of confidence 
intervals are given; one involves using log transformations and another using the delta-theorem (Casella and Berger 
(1990)).  Currently, it is unclear whether one interval is superior to another, but simulation seems to point toward 
back-transformed intervals having better coverage probabilities.  The SAS macros which calculate these estimates 
give both sets of confidence intervals as well as a standard error estimate from the delta-theorem based confidence 
intervals.

DOUBLY ROBUST ESTIMATE
The primary issue with using the regression based estimate of attributable benefit is that it can be potentially biased 
when the outcome regression model is misspecified.  While it is certainly true that our estimates rely on an 
assumption of no unmeasured confounders, a violation of which certainly has the potential to introduce biases. 
Alternatively, suppose that all the confounders are measured appropriately but the outcome regression model is 
misspecified due to leaving out important terms like transformations or interactions.  This type of misspecification can 
also lead to biased estimates of AB.

Before going any further it is important to note that the idea of misspecification really takes two forms here.  First, the 
optimal treatment regime could be misspecified.  If we estimate that the optimal method of treating patients is to give 
everyone the standard treatment, when in reality the best treatment option would include a subset of patients who 
should receive the alternate treatment.  There is really little we can do in such cases because the misspecification is 
not on the public health benefit, but on what is indeed the best strategy to treat patients.  Suppose on the other hand 
that we have correctly specified the optimal treatment regime, potential biases still linger if the outcome regression 
model is misspecified.  The purpose of a doubly robust estimate of attributable benefit is to offer an additional layer of 
protection from this type of misspecification.  

So in keeping with this idea let’s will start by assuming that the optimal treatment regime is indeed as defined by our 
function gopt.  That is to say that we believe that the algorithm for treating patients is the best method for assigning 
treatments.  Now the idea is to have an estimate of that regime’s attributable benefit that is free from potential biases 
due to model misspecification.  We do this by introducing a doubly robust estimate of attributable benefit, which 
combines the outcome regression model with a propensity score model and the use of inverse probability weighting. 
The key is to augment the current estimate using a second model that looks at the propensity toward receiving 
certain treatments.  Suppose we posit that our data comes from the true propensity score model

In addition define C(gopt) as a function that indicates when a patient’s observed treatment is in line with the optimal 
treatment regime.  That is let

and we can estimate a propensity model for C via

where π is our estimate of the propensity score model easily accessible in SAS via Proc Logistic.  From this we can 

define another estimate of ABopt as 

where the function hdr is

For those unfamiliar with doubly robust estimates, the above would seem complicated and difficult to interpret.  The 
interested reader should see Faries et al. (2010) for an excellent overview of doubly robust estimation and its’ 
implementation in SAS.  For this particular example the function hdr is calculated in a series of steps; find gopt using 
the outcome regression model, then calculate C by determining the cases when patients received the treatment that 
is in line with the policy gopt.  Next fit the propensity model for treatment and calculate vopt which is the estimated 
propensity of the observed treatment coinciding with treatment policy gopt.  Using these elements and the estimated 
probabilities from the outcome regression model, one can now estimate hdr.

In practice it is important for those unfamiliar with doubly robust estimation that there is generally a trade-off in 
efficiency to ensure unbiased estimates.  Therefore it is common in small samples that the regression estimate of 
attributable benefit to have a smaller confidence interval than it’s doubly robust counterpart.  The analyst should 
decide apriori whether misspecification may be an issue and select the appropriate macro beforehand.  In addition, 
the above estimate uses inverse propensity weights which can be problematic when the propensities are very small. 
In a situation where the propensity toward treatment is 0.01, the inverse weight would count that observation 100 
times which can also lead to potential biases.

Again it can be shown that the log of 1 minus the above estimate of attributable benefit is asymptotically normal using 
the theory of influence functions, currently this is a matter of ongoing research.  There is an added layer of difficulty in 
variance estimation due to the use of multiple statistical models, the influence functions must be calculated using 
matrix algebra.  Add to this the fact that the above estimate uses indicator functions for C and vopt and that the 
estimated optimal treatment regime is dependent on the estimated model parameters, then an issue arises because 
there is no closed form derivative to use in a sandwich variance estimate.  This issue of a closed form derivative is 
also an issue in variance estimation of the regression based estimate and is solved in the macros by using a simple 
numeric derivative.

SAS MACROS FOR ABopt

While the underlying theory for estimating attributable benefit is complex, the implementation is quite straight forward. 
Let us start by assuming the analyst has found the “best” propensity and outcome regression logit models on their 
own.  The beauty here is that the analyst may use whatever model selection technique they deem most appropriate. 
Also, simulation seems to suggest that a slightly overfit model has relatively little impact.    

To use the macro for a regression based estimate one only needs to discern the best outcome regression model. 
The macro for calculating this estimate of ABopt can be found online at https://www.ecu.edu/cs-dhs/bios/ABopt.cfm or 
in the appendix of this manuscript.  Overall the initial outcome regression code should be similar to the following:

Proc Logistic;
Class T;
Model Y = T X;.

Here only one covariate is listed but the macro is intended for multiple confounding variables.  If categorical variables 
are to be used in the model, the analyst should make the indicator functions on his or her own instead of letting Proc 
Logistic do it.  Inputs for the outcome regression macro are 

%AB_opt_reg(alpha, Data, Y, T, X, X2, Int, Out, Interaction);
where alpha is the significance level, Data is the name of SAS dataset to be used, Y is the outcome variable, T is 
treatment variable, X is one or more continuous or binary covariates.  X2 is the input for those variables in X that have 
significant interactions with treatment, while Int is the input for analyst created interaction variables between 
exposure and treatment.  Out is the name of a dataset for which SAS can output the estimate of ABopt, standard 
error, and associated confidence intervals.  Interaction is 0 if there is no treatment by covariate interaction and 1 
if there is treatment by covariate interaction.

The inner workings of the macro are straight-forward, Proc Logistic is used in the manner as described above.  ODS 
is used to output the model parameter estimates table.  Those parameter estimates and the original data are then 
called in SAS IML for matrix calculations for estimating for ABopt, standard errors, and both sets of confidence 
intervals.  In general, the back-transformed confidence intervals usually provide better coverage but if one needs a 
standard error estimate of ABopt then the corresponding delta-theorem intervals may be more appropriate.

The macro for the doubly robust estmate of ABopt works in a similar fashion; however, in addition to an outcome 
regression model specified above, a propensity model is also fit similar to the following:

Proc Logistic;
Model T = X;.

The syntax for this macro is the same as above and the macro command is 
%AB_opt_DR(alpha, Data, Y, T, X, X2, Int, Out, Interaction);

where the inputs are the same as listed above.  Again, parameter estimates for both the outcome regression and 
treatment propensity models are output via ODS.  Those estimates along with the original data are sent to Proc IML 
for variance and confidence interval calculations.  The results of which are saved in a separate dataset specified by 
the user in the macro.  While this second macro is not currently in public circulation, interested parties may request a 
copy via e-mail provided it is for limited use.

EXAMPLE
As an example to illustrate the utility of both macros, consider a hypothetical example of data generated via the 
following code:

Data Sample;
i=1;
Do while (i<1001);
X=rannor(123);
Error = .5*rannor(2);
Error2= .5*rannor(3);
*propensity model;
m =exp(-1.25-X+Error);
Pi_X=m/(1+m);
T = ranbin(1111,1,Pi_X);
XT = X*T;
*outcome regression model;
m2=exp(-3  + .25*T + 1.1*X -1.5*XT + Error2);
mu = m2/(1+m2);
Y = ranbin(2222,1,mu);

if pi_x > .01 then output;
if pi_x > .01 then i=i+1;
end;
keep Y T X XT; 

The above code generates 1000 observations of the form (Y,T,X), note that any constellation of parameter estimates 
would work to illustrate these methods.  In this example, X is distributed N(0,1), P(T=1) is approximately 0.27, and 
P(Y=1) is approximately 0.09. Again, the bound on the propensity model is proposed only to ensure that biases that 
arise from inverse propensity weight based estimators are avoided.  Here we are stipulating that all patients are 
estimated to have at least a 1% chance of receiving the alternate treatment.  Since the data are generated with noise, 
our parameter estimates will coincide with, but not be exactly equal to those specified in data generation.  In fact the 
propensity model parameter estimates are

Analysis of Maximum Likelihood Estimates

Parameter DF Estimate Standard
Error

Wald
Chi-Square

Pr > ChiSq

Intercept 1 -1.3844 0.0864 256.4802 <.0001

X 1 -0.8764 0.0916 91.6342 <.0001

and the outcome regression model parameter estimates are

Analysis of Maximum Likelihood Estimates

Parameter DF Estimate Standard
Error

Wald
Chi-Square

Pr > ChiSq

Intercept 1 -3.1107 0.2056 228.9433 <.0001

T 1 -0.1177 0.4573 0.0662 0.7969

X 1 1.3014 0.1668 60.8562 <.0001

XT 1 -1.9701 0.3265 36.4043 <.0001

hence there is a good deal of noise in the actual parameter estimates.  Had this been actual data our conclusions 
would be that the alternate treatment is the most beneficial for the majority of patients.  The estimated optimal 
treatment regime would be to give all patients the alternate treatment unless they have an extremely high positive 
value of X, in which case the standard treatment is more beneficial.  From here it is still unclear exactly what the 
overall impact of treatment is, hence estimating ABopt will attempt to quantify this.

Starting with the regression based macro, our inputs for a proper estimate of ABopt is 
%AB_opt_reg(0.05, Sample, Y, T, X, X, XT, AB_Out, 1);.

Reading the inputs from left to right, our significance level is 0.05, the input dataset is Sample, outcome and 
treatment variables are Y and T, X is a covariate, X is also part of an interaction effect which we have created and 
called XT, the output dataset with our results will be called AB_Out, and there is indeed a treatment by covariate 
interaction in the model.  From this code the estimate for the probability of a poor outcome using the optimal 
treatment regime is 0.0221 and ABopt  is estimated at 0.7301.  This implies that roughly 73% of the poor outcomes 
could have been prevented had the optimal treatment strategy been implemented.  Back-transform confidence 
intervals are (0.5459, 0.8396) while delta-theorem based confidence intervals are (0.5897, 0.8706).  The delta 
theorem standard error is 0.0716.

By comparison, the macro for a doubly robust estimation has the same inputs as above is 
%AB_opt_DR(0.05, Sample, Y, T, X, X, XT, AB_Out, 1);.

For this example, the doubly robust estimate of ABopt is slightly higher at 0.7873.  The back-transform confidence 
interval (0.4932, 0.9107) and the delta-theorem based interval (0.6026, 0.9720) are wider than the regression based 
interval.  The delta-theorem standard error for this estimate is 0.0942 roughly 30% higher than the standard error of 
the outcome regression based estimate.

In order to see the utility of the doubly-robust estimate, we must examine the impact under model misspecification. 
Consider the same scenario, but suppose the interaction term was left out of the model.  Since there is no longer an 
interaction effect to the model, the optimal treatment strategy will be to give all patients the same treatment.  Recall 
that we refer to this strategy as ABdom and in scenarios where there is no treatment by covariate interaction, ABdom = 
ABopt.

Changing the macro inputs for a misspecified model that leaves out the interaction term would look like 
%AB_opt_reg(0.05, Sample, Y, T, X, , , AB_Out, 0);.

Note that the propensity model would not change in the doubly robust macro, but the outcome regression model 
estimates would change to the following:

Analysis of Maximum Likelihood Estimates

Parameter DF Estimate Standard
Error

Wald
Chi-Square

Pr > ChiSq

Intercept 1 -2.7520 0.1647 279.1645 <.0001

T 1 0.3343 0.3190 1.0979 0.2947

X 1 0.8696 0.1353 41.3275 <.0001

Note that there is a non-significant treatment effect.  The regression based estimate of ABopt (ABdom) is 0.0482 with 
back-transformed confidence interval (-0.0417, 0.1302).  While the doubly robust estimate of ABopt is 0.079, almost 
double the estimated benefit from the regression based estimate, with back-transformed confidence interval (-0.0232, 
0.1710).  From this there are two important items; first, the confidence intervals for both estimates agree with the 
conclusion made from the model parameter estimates (there is no significant treatment impact).  Second, a 
misspecification here was on both the optimal treatment regime and that regime’s attributable benefit.  While there is 
nothing we can do to protect from optimal regime misspecification, the doubly robust macro provides a better 
estimate of the specified regime’s public health impact.  Note that in practice, the analyst should be able to 
demonstrate a clear impact of either treatment before trying to assess the benefit of any one particular treatment 
strategy.  Some of the underlying theory require certain regularity conditions which can go awry when neither the 
standard nor alternate treatment have any public health impact.

CONCLUSION
From the example one can really start to see the potential utility of a measure like attributable benefit as well as both 
estimating macros.  AB is powerful in that it can put together all the information about the public health impact of 
various treatment strategies.  This work is a staging point from which different treatment regimes (both dynamic and 
static) can be compared to help determine a total cost/benefit framework.  The regression based estimate of ABopt is 
easy to use and can handle any number of possible covariates.  The doubly robust estimate of ABopt is a viable 
alternative in cases where it is unclear where the outcome regression model is specified correctly.  There is an 
obvious trade-off between bias and efficiency, but this can be valuable in many community health settings.  The 
analyst generally has access to those making treatment decisions and can gain a better understanding of which 
factors go into making those decisions.  Hence a proper dialogue between analyst and physician is more likely to 
produce a properly specified propensity model than an outcome regression model that is put together with a set of 
assumptions and hypotheses. 

The ideal way for the macros here to be used would be at the end of all other analyses, after one selects the best 
model for treatment and propensity.  While the example above demonstrated the effects of model misspecification, 
the underlying assumption of no unmeasured confounding should imply that the analyst build the most accurate 
model as possible before estimating these type of effects.  

There is enormous room for future work in this area, the current macros are very limited and extensions need to be 
made for treatment strategies involving more than two treatments, continuous or survival-type outcomes, time-
dependent treatment regimes, and macros that find the attributable benefit of any specified treatment strategy (either 

dynamic or static).  In addition, the large sample variance estimates are quite cumbersome and it is possible that 
bootstrap standard errors and confidence intervals may be as effective as those used here. Estimating the optimal 
treatment regime is a continued subject for ongoing research and hopefully later versions of these macros will be able 
to incorporate some of those methods.

APPENDIX
SAS Macro for regression based estimate of the attributable benefit of the optimal treatment regime:
/*******************************************************************************
|
| Program Name:  AB opt covariate simulation
|
| Program Version: 1.0
|
| Program Purpose:  Calculates AB for the optimal treatment regime from a 
|  prespecified model.  User must preload data into SAS work folder.
|                 
| SAS Version: 8 or 9
|
| Created By:   Jason Brinkley
| Date:            19-Feb-2009
|

|*******************************************************************************
| Change Log
|
| Modified By:
| Date of Modification:
|
| Modification ID:
| Reason For Modification:
|
*******************************************************************************/

/*Instructions:
Load data into SAS work folder. Data must be of form: Binary Disease, Binary
Exposure/Treatment, Continuous or Binary Covariates.  For discrete variable input,
analyst needs to make binary indicators for each level of covariate and input those
binary indicators instead of discrete variables.  User must create interaction
variables in the dataset before running macro.  Analyst needs to use other
techniques to find "best" model before running macro.

Macro inputs are as follows:
alph - Confidence level for intervals (i.e. 95% intervals imply alpha=0.05)
Data - name of dataset in work folder
Y -  Binary outcome/disease variable
T -  Binary exposure/treatment variable
X -  one or many covariates.  Continuous or Binary only.
Interactions between covariates go here.
X2 -  Which covariates/exposure interactions are included in the model
(i.e. if T*X is significant then put X here)
Int -  User created interaction variables in model
(i.e. if T*X is significant, analyst creates variable TX=T*X
  TX goes here)
Out -  Name of output dataset for further manipulation
Interaction -  Indicator of whether there is covariate/exposure interactions
in the model (0 = no interactions, 1 = interactions). 
IF THERE ARE COVARIATE/COVARIATE INTERACTIONS ONLY THEN PUT 0 HERE.*/

%Macro AB_opt_reg(alph, Data,Y,T, X, X2, Int, Out, Interaction);
*Different logit models whether there is interactions;

%IF &Interaction = 0 %Then %Do;
proc logistic descending data=&Data;
model &Y = &T &X;
ods output ParameterEstimates = ParameterEstimates;
run;
%End;

%IF &Interaction = 1 %Then %Do;
proc logistic descending data=&Data;
model &Y = &T &X &Int ;
ods output ParameterEstimates = ParameterEstimates;
run;
%End;
quit;

Data Betas;
set ParameterEstimates;
Keep Data Variable Estimate StdErr;
run;

proc iml;
*Output files step, initializing variables;
P_Yopt=0;

P_Y=0;
Lower_BT=0;
Upper_BT=0;
AB_opt_hat = 0;
ln_ratio = 0;
lower_DT =0;
upper_DT = 0;
se_ab=0;

*Load data;
use &Data;
read all var {&X} into X;
read all var {&T} into T;
read all var {&Y} into Y;
read all var {&X2} into X2;
if &Interaction = 1 then read all var {&Int} into int;
*Calculate estimates for P(Y=1);
P_Y=(sum(Y))/(nrow(Y));
IF3=(Y-P_Y)/P_Y;
Use Betas;
read all var{estimate} into B;
read all var{stderr} into StdErrB;

*bound needed for numeric derivatives;
bound = .01 * StdErrB;
bound2= bound;
r = nrow(B);
n=nrow(Y);
Outvar = n||P_Yopt || P_Y || AB_opt_hat ||ln_ratio || Lower_BT ||Upper_BT
 ||Lower_DT || Upper_DT ||SE_AB  ;
 cname = {"Sample Size" "Prob Y optimal" "Prob Y" "AB opt hat"
 "ln(ratio)" "Lower BT" "Upper BT" "Lower DT" "Upper DT" "Delta SE" };
 create out from Outvar [ colname=cname ];
*f, f_0, and f_1 are different whether there are interactions or not;
I=j(n,1);
if &Interaction=0 then f=T(I||T||X);
if &Interaction=1 then f=T(I||T||X||Int);
I1=j(n,1);
*I2 is a vector of zeros;
I2=I1-I1;
if &Interaction=0 then f_0=T(I1||I2||X);
if &Interaction=1 then f_0=T(I1||I2||X||I2);
if &Interaction=0 then f_1 = T(I1||I1||X);
if &Interaction=1 then f_1 = T(I1||I1||X||X2);
T_opt = T;
*create T_opt, T_opt chooses level with lowest chance of poor outcome. 
If the chance of a poor outcome is the same then macro defaults to treatment 1;
do k = 1 to n;
T_opt[k,] = 0;
M1 = exp(T(B)*f_0[,k]);
M2 = 1/(1+exp(T(B)*f_0[,k]));
M=M1*M2;
W1 = exp(T(B)*f_1[,k]);
W2 = 1/(1+exp(T(B)*f_1[,k]));
W=W1*W2;
if W < M then T_opt[k,]=1;
end;
*X2 may be a subset of X variables;
if &Interaction=1 then ToptX = T_opt#X2;

if &Interaction=0 then f_opt = T(I1||T_opt||X);
if &Interaction=1 then f_opt = T(I1||T_opt||X||ToptX);

avg1 =0;
*P_Yopt calculation and IF1 calculation;
M1 = exp(T(B)*f_opt);
M2 = 1/(1+exp(T(B)*f_opt));
M=M1#M2;
avg1=M[,:];
IF1=M-avg1;
IF1=T(IF1);
P_Yopt=avg1;
*IF2 Calculation;
Avg2=0;
ncol = nrow(f);
Temp = I(ncol);
*avg is a matrix of zeros;
Avg3=T(Temp - Temp);
*numeric derivatives;
row_vector = j(1,r,1);
LowB = B*row_vector;
UpB = LowB;
*need these for the numric derivatives later;
do v = 1 to r; 
LowB[v,v] = B[v,]-bound[v,];
UpB[v,v] = B[v,]+bound[v,];
bound2[v,]=.5*(1/bound[v,]);
end;
*numeric Derivative loop;
ND1 = B;
ND2 = B;
Do v = 1 to r;
NM1 = exp(T(LowB[,v])*f_opt);
NM2 = 1/(1+exp(T(LowB[,v])*f_opt));
NM3 = sum(NM1#NM2)/n;
ND1[v,1] = NM3;
NM4 = exp(T(UpB[,v])*f_opt);
NM5 = 1/(1+exp(T(UpB[,v])*f_opt));
NM6=sum(NM4#NM5)/n;
ND2[v,1]= NM6;
end;
*ND is numeric approximation for partial mu/partial beta;
ND = (ND2 - ND1)#bound2;
Do k = 1 to n;
Q1=f[,k]*T(f[,k]);
Q2 = exp(T(B)*f[,k]);
Q3 = (1+exp(T(B)*f[,k]))*(1+exp(T(B)*f[,k]));
Q3=1/Q3;
Q4=(Q2*Q3)*Q1;
Avg3 = Avg3 + Q4;
end;
Avg3 = Avg3/n;
Avg3 = inv(Avg3);
IF2 = T(ND)*Avg3*f;
IF2 = T(IF2);
R1 = exp(T(B)*f);
R2 = 1/(1+exp(T(B)*f));
R=T(R1#R2);
IF2 = IF2 # (Y-R);
*Put all the pieces together to get IF;
IF = ((IF1+IF2)/p_Yopt)-IF3;
V = T(IF) * IF;
V = V / ((n-1)*(n-1));
V = sqrt(V);
*V is estimate of the standard error of
ln(PYstar0=1)-ln(P Y=1) now need to put back together;

Ratio=P_Yopt/P_Y;
AB_opt_hat = 1-ratio;
ln_ratio =log(ratio);
alph= &alph ;
z = probit(1-(alph/2));

*Back Transform 95% Confidence Interval;
Lower_BT =1-exp(ln_ratio + (z* V));
Upper_BT=1-exp(ln_ratio - (z*V));

*Delta Theorem 95% Confidence Interval;
Lower_DT = AB_opt_hat - (z*ratio*V);
Upper_DT = AB_opt_hat + (z*ratio*V);
SE_AB = ratio*V;

*Output needed values;
Outvar = n || P_Yopt || P_Y || AB_opt_hat ||ln_ratio || Lower_BT ||
Upper_BT ||Lower_DT || Upper_DT ||SE_AB ;
append from Outvar;

********************************************************************************
*
* Start of analysis for output
*
*******************************************************************************;

Data &out;
set out;
label
Sample_Size = 'Sample Size'
ln_ratio_ = 'Log{P(Y)/P(Y_opt)}'
Prob_Y_optimal = 'Probability Y optimal'
Prob_Y = 'Probability Poor Outcome'
AB_opt_hat = 'AB opt hat'
Lower_BT = 'Backtransformed Lower 95% C.I.'
Upper_BT = 'Backtransformed Upper 95% C.I.'
Lower_DT = 'Delta Thm Lower 95% C.I.'
Upper_DT = 'Delta Thm Upper 95% C.I.'
Delta_SE = 'Delta Thm Standard Error';

Proc Print data=Out label;
run;

quit;
quit;

%MEND;

