

How To Use Proc SQL select into for List Processing

Ronald J. Fehd, Centers for Disease Control and Prevention, Atlanta, GA, USA

ABSTRACT

Description : The SAS® macro language is simple, yet powerful. List Processing with Proc SQL is also simple, yet powerful. This Hands On Workshop paper provides programmers with knowledge to use the Proc SQL select into clause with the various SQL dictionaries to replace macro arrays and %do loops.

Purpose : list processing, review sql dictionaries, writing text with sql

Audience : intermediate to advanced users, and macro programmers.

Keywords : dynamic programming, list processing, macro, SQL

Information : writing alphabetical list of macro variables, processing items in list

Contents

Introduction	2
Basics and Concept Review	3
SQL Basic Processes	5
Writing Constant Text	8
Dictionary Dictionaries	11
Dictionary Macros	13
Dictionary Options	15
Writing Macro Calls	17
Macros	18
Dictionary Columns	20
Dictionary Tables	23
Filenames	25
Using Macro %Do Loops	27
Using Items in Delimited List	28
Using Arrays of Macro Variables	30
Conclusion	32
Suggested Readings	33
Bibliography	34
Index	38

Introduction

Overview

This paper reviews the theory of programming, how to process every item in a list using the simplicity of Proc SQL select into list processing with SQL's dictionary tables. It provides basic answers to these questions:

1. How do I process every column in a dataset?
 2. How do I process every file in a folder?
 3. How do I process every member in a libref?
- or*
4. How do I process every *item* in a *list*?

Prerequisites

Students are expected to have the following minimum background:

programming : three to seven years
data step : data structure allocation
with attribute or length statements
macro language : allocate macro variables
write macros with one or more steps
procedures : Contents, Print

Topics

This paper cover the following topics:

- programming theory: vocabulary
- Proc SQL syntax
- list processing (dynamic programming) with dictionaries:
 - columns
 - dictionaries, v9
 - filenames, not an sql dictionary, read with scl functions
 - macros
 - options, v9: group
 - tables
- v9 concatenation functions: catt, catx

Basics and Concept Review

Overview

This section provides background information.

Programming Theory and Vocabulary

We communicate in a natural language English (or Chinese, Dutch, French, or German) about the artificial language SAS. These computer science terms and concepts are used throughout this paper.

Terms Used Here		
program	data structure algorithm	(also: metadata)
data structure	attributes	declarative, information (compile-time) statements name : variable or column type : character or numeric length: in bytes character: 1–32,767 numeric : 1– 8 format label see Online Help: Index: declarative DATA step statements organization executable: action or control statements array: has numbered elements do I=1 to dim(array-name); put array-name(I); list : <i>has unnumbered items</i> do over array-name; put array-name; drop, keep retain where
algorithm	input: process output:	libref, data, variable perform actions libref, data, to log or list

SQL syntax

There are five SQL statements:

1. proc
2. create
3. describe
4. select
5. quit

```
1 PROC SQL;
2 PROC SQL noprint;
3     create    table table-name as
4               query-expression
5               <order by order-by-item
6               <, ... order-by-item>>;
7     describe table table-name <,> ... table-name>;
8     select    <distinct> object-item
9               <function>(object-item)
10              <,> ... object-item>
11              into :macro-variable <separated by ' '>
12                  :macro-variable-A, :macro-variable-B
13                  :macro-variable1 - :macro-variable&SysMaxLong.
14     from      Libref.Data
15     where     ColumnChar eq 'value'
16             and ColumnChar2 eq 'value2'
17             or ColumnNum eq <num-value>
18     group by  group-by-item
19             <,> ... group-by-item>>
20     having    sql-expression
21     order by  order-by-item
22             <,> ... order-by-item>>;
23     quit;
```

The keyword `select` has one required clause, `from`, and five optional clauses: `into`, `where`, `group by`, `having`, and `order by`, which may be viewed conceptually in this hierarchy:

```
select
  into
  from
    where
  group by
    having
  order by
```

SQL Basic Processes

Overview

Proc SQL can be used to do each of the basic processes:

- list data structure
- list data
- list only subset
- create data
- make unique list

Listing Data Structure

Proc SQL produces a data structure listing in the log like Proc Contents does in the listing. Instead of the `data = option`, SQL has the `describe table statement`.

```
ProcSQL-1-describe-table-libref-data.sas
1 PROC Contents data = SAShelp.Class;
2
3 PROC SQL; describe table SAShelp.Class;
4 quit;
```

Listing Data

Proc SQL works like Proc Print. In the Print method the object is referred to with the `data = option`. In SQL the object reference is the `from clause`. Star — asterisk (*) — means all variables.

```
ProcSQL-1-select-star.sas
1 PROC Print data = SAShelp.Class;
2 var _all_;
3
4 PROC SQL; select * /* _all_ */
5 from SAShelp.Class;
6 quit;
```

Listing Subsets

The `where` statement is available in all procedures. This shows it as a data set option. The SQL `select . . . from statement` has a `where` clause.

```
ProcSQL-1-select-subset.sas
1 PROC Print data = SAShelp.Class
2 (where = ( Sex eq 'F'
3 and Age ge 14));
4 var Name Age;
5
6 PROC SQL; select Name, Age
7 from SAShelp.Class
8 where Sex eq 'F'
9 and Age ge 14;
10 quit;
```

Creating Data

A common task is to copy a permanent data set from a permanent storage library to the work library. The SQL statement `create table` provides a similar data manipulation environment.

```
ProcSQL-1-create-table-libref-data.sas
1 DATA    Work.Class;
2 set     SAShelp.Class;
3
4 PROC SQL; create table    Work.Class as
5             select *
6             from    SAShelp.Class;
7             quit;
```

Making Unique List

The SQL `select` statement has a `distinct` function, which can be used to collapse many instances of variable values into a unique list.

```
ProcSQL-1-unique.sas
1 PROC Sort data  = SAShelp.Class
2           (keep  = Age)
3           nodupkey
4           out   = Work.UniqueAge;
5           by    Age;
6
7 PROC SQL; create table UniqueAge as
8           select distinct Age
9           from    SAShelp.Class
10          quit;
```

Summary of Basic SQL

There are several differences between the syntax of `proc SQL` and other procedures.

The two most important to note are that column (variable) names are separated by commas, and dictionary tables' values are upper case.

select : use comma as delimiter between column names

wrong : `select Column1 Column2 Column3`

right : `select Column1, Column2, Column3`

where : values in dictionary tables are upper case

wrong : `where Libname eq 'SAShelp'`

right : `where Libname eq 'SASHELP'`

References

These papers provide an introduction to SQL: Hermansen [26, sugi22.035] Ronk [34, sugi29.268] Wells [39, sugi26.105] Winn, Jr. [43, sugi22.067]

Concatenation Functions

The v9 concatenation functions are used here to replace the double bang (!!), also known as exclamation point, or double vertical bar (||).

Program : This program compares the concatenation operator and cat* functions.

```

1  DATA Work.Cat_functions;
2  attrib Verb    length = $16
3         Text    length = $32
4         Prefix  length = $ 8
5         Infix   length = $ 8
6         Suffix  length = $ 8
7         One     length =  4
8         Two     length =  4
9         Three2  length =  8;
10 Prefix = 'prefix- ';
11 Infix   = ' *infix* ';
12 Suffix  = ' -suffix';
13 One     = 1;
14 Two     = 2;
15 Three2  = 3.2;
16 Verb = 'concat';
17 Text = Prefix !! Infix !! Suffix ;
18 output;
19 Verb = 'concat trim';
20 Text = trim(Prefix) !! trim(Infix) !! trim(Suffix) ;
21 output;
22 Verb = 'concat left trim';
23 Text = trim(left(Prefix)) !! trim(left(Infix))
24       !! trim(left(Suffix)) ;
25 output;
26 Verb = 'cat';
27 Text = cat(Prefix,Infix,Suffix) ;
28 output;
29 Verb = 'catt';
30 Text = catt(Prefix,Infix,Suffix) ;
31 output;
32 Verb = 'cats';
33 Text = cats(Prefix,Infix,Suffix) ;
34 output;
35 Verb = 'catx';
36 Text = catx('+',Prefix,Infix,Suffix) ;
37 output;
38 Verb = 'catx numbers';
39 Text = catx('+',One,Two,Three2) ;
40 output;
41 stop;
42 Proc Print;

```

Listing : Notice the presence of double spaces in obs one and four.

```

1  Obs    Verb    Text
2
3  1      concat  prefix-  *infix* -suffix
4  2      concat trim  prefix- *infix* -suffix
5  3      concat left trim  prefix-*infix*-suffix
6  4      cat      prefix-  *infix* -suffix
7  5      catt     prefix- *infix* -suffix
8  6      cats     prefix-*infix*-suffix
9  7      catx     prefix-+*infix*+-suffix
10 8      catx numbers  1+2+3.2

```

Writing Constant Text

Overview

SQL provides the ability to select text as well as columns in its queries. In this section we examine first a simple example of selecting text and then see how to place both values from columns and surrounding text which comprise complete SAS statements into a macro variable, which is used to submit the text for processing.

In this section:

Dictionary Dictionaries	11
Dictionary Macros	13
Dictionary Options	15

Selecting Text

The `select` statement accepts strings as one of its arguments; each string can be either single- or double-quoted, which allows the use of macro variables.

```
_____ ProcSQL-2-select-constant-text.sas _____
1  %Let Rows = rows;
2
3  PROC SQL; select MemName, 'has', Nobs, "&Rows."
4                from Dictionary.Tables
5                where LibName eq "SASHELP"
6                      and MemName eq "CLASS"
7                      and MemType eq "DATA";
8                quit;
```

Note: the length of column `MemName` is 32, which accounts for the wide space between the words `Class` and `has`.

```
_____ ProcSQL-2-select-constant-text.sas _____
1
2                Number of
3                Physical
4  Member Name      Observations
5  -----
6  CLASS                        has      19 rows
```

Creating a Macro Variable

This example shows how to concatenate concatenate text and variable value, using the `catt` (concatenation, trimmed) function and put that text into a macro variable.

For each row the statement is:

```
Proc Contents data = SASHELP.MemName;
```

```
_____ ProcSQL-2-select-text-into-List.sas _____
1  %Let List = *missing;* initialize for no rows selected;
2
3  PROC SQL; select catt('Proc Contents data = SASHELP.'
4                    ,MemName
5                    ,';run;'
6                    )
7                into :List separated by ' '
```



```

8      from Dictionary.Tables
9      where LibName eq "SASHELP"
10         and MemName like "V%"
11         and MemType eq "VIEW";
12      quit;
13 &List. ; * execute statements in mvar List;
14

```

Notes:

- into :** The `into` clause creates the macro variable named after the colon. Notice the the last argument of the `catt` function is a semicolon and that the delimiter of `separated by` is space. This could also be:
`catt(...,Memname) ...separated by ' ; '.`
- like :** The `like` operator chooses only names beginning with 'V'; these tables correspond to sql dictionaries.
- quit :** The statements in the macro variable are procedure statements, therefore they must be executed *after* the `quit` statement.

List Processing Statements

These statements are in the macro variable List.

```

_____ ProcSQL-2-select-text-into-List.lst, snip 1 _____
1 Proc Contents data = SASHELP.VALLOPT;
2 Proc Contents data = SASHELP.VCATALG;
3 ...
4 Proc Contents data = SASHELP.VVIEW;

```

Program Output

Output from statements in the macro variable List.

```

_____ ProcSQL-2-select-text-into-List.lst, snip 2 _____
1 The CONTENTS Procedure
2
3 Data Set Name      SASHELP.VALLOPT      Observations      .
4 Member Type       VIEW                Variables         6
5 Engine            SQLVIEW              Indexes           0

```

References

SASHELP views associated with SQL dictionaries are reviewed by Dilorio and Abolafia [15, sugi29.237].

Compare with program `ProcSQL-D-Dictionaries-list-describe-table` below, which lists SQL dictionaries.

List Processing Summary

There are several steps in writing statements to a macro variable and executing them:

1. input: data structure
 - (a) identify the input table
 - (b) examine its data structure
2. process: expected text
 - (a) concatenation of text and values into macro variable
 - i. clarify the text surrounding the variable value(s):
prefix, infix(es), suffix
 - ii. identify subset, if any; are values in ALL CAPS?
 - iii. volume of text in macro calls: can information be gotten by the subroutine macro rather than passed as a parameter? this caveat is especially important when passing long character variables such as data set or variable labels.
see: ¶ problems below
 - (b) remember closure or delimiter:
clause(s): comma, space, or other
statement(s): semicolon (;)
step boundary: `run;`
 - (c) execute the statements: SQL: *before* `quit;`
SAS: procedures, macros: *after* `quit;`
3. output: consider ODS

In the next sections we use this list processing check list to examine several of the more commonly used dictionaries.

Commentary

The first examples in the next section — dictionaries, macros and options — illustrate writing constant text. In the second section we look at how to write macro calls to generate more complicated amounts of text when reading columns (variables), filenames, and tables.

Dictionary Dictionaries

Overview

SQL provides a meta-dictionary, Dictionary.Dictionaries, which contains a description of all other dictionaries.

Can we use this meta-dictionary to write statements which describe all the other dictionaries?

Problem Statement

The task is to write these statements:

```
describe table Dictionary.MemName;
```

Dictionary.Dictionaries is unique on MemName and Name (Column Name).

We need to make a data set (list) unique on MemName.

Review Data Structure

The describe table statement lists the data structure.

Note: V9 only.

```
1 _____ ProcSQL-D-Dictionaries-describe-table.sas _____
2 PROC SQL; describe table Dictionary.Dictionaries;
3 quit;
```

The data structure is written to the Log:

```
1 _____ ProcSQL-D-Dictionaries-describe-table.log _____
2 create table DICTIONARY.DICTIONARIES
3 (
4   memname char(32) label='Member Name',
5   memlabel char(256) label='Dataset Label',
6   name char(32) label='Column Name',
7   type char(4) label='Column Type',
8   length num label='Column Length',
```

Program

This program makes a data set with the names of all the SQL dictionaries using select distinct function, writes describe table statements for each, then executes those statements. The log contains the data structure of each SQL dictionary.

```
1 _____ ProcSQL-D-Dictionaries-list-describe-table.sas _____
2 *note: need v9 for D.Dictionaries;
3 Proc SQL; create table List as
4   select distinct MemName
5   from Dictionary.Dictionaries;
6   select catt('describe table Dictionary.'
7   , MemName
8   , ';'
9   )
10  into :List separated by ' '
11  from List ;
12  &List.;
13  quit;
14  run;
```

The statements in the macro variable are SQL statements, therefore they must be executed *before* the `quit` statement.

List Processing Statements

These are the statements in the macro variable List:

```
ProcSQL-D-Dictionaries-describe-table.lst
1 describe table Dictionary.CATALOGS;
2 describe table Dictionary.CHECK_CONSTRAINTS;
3 ...
4 describe table Dictionary.VIEWS;
```

Program Output

SQL table descriptions are written to the log.

```
ProcSQL-D-Dictionaries-describe-table.log
1 NOTE: SQL table DICTIONARY.CATALOGS was created like:
2 ...
3 NOTE: SQL table DICTIONARY.CHECK_CONSTRAINTS was created like:
4 ...
5 NOTE: SQL table DICTIONARY.VIEWS was created like:
```

Commentary

Compare with program `ProcSQL2-select-text-into-List` above, which lists SAShelp views.

This program illustrates a two-statement solution to this problem. See program `ProcSQL-D-Options-list-groups` below for a single-statement solution.

Dictionary Macros

Overview

SQL maintains a list of all macro variables.

One problem with global macro variables is checking their values.

The statement `%Put _user_` writes an unsorted list of macro variable names and values to the log.

```
_____ Put-User.log _____
1      %Let A = 1;
2      %Let b = 22;
3      %Let z = end of list;
4      %Put _user_;
5 GLOBAL Z end of list
6 GLOBAL A 1
7 GLOBAL B 22
```

Problem Statement

The task is to write a `%put` statement for each macro variable. Hard-coded this would be:

```
%put a: &a.;
%put b: &b.;
%put z: &z.;
```

Review Data Structure

These statements list the data structure of Dictionary.Macros:

```
_____ ProcSQL-D-Macros-describe-table.sas _____
1 PROC SQL; describe table Dictionary.Macros;
2       select  distinct Scope
3       from    Dictionary.Macros;
4       quit;
5 run;
```

Here is the data structure:

```
_____ ProcSQL-D-Macros-describe-table.log _____
1 create table DICTIONARY.MACROS
2 (
3     scope char(32) label='Macro Scope',
4     name char(32) label='Macro Variable Name',
5     offset num label='Offset into Macro Variable',
6     value char(200) label='Macro Variable Value'
```

The values of scope are:

```
_____ ProcSQL-D-Macros-describe-table.lst _____
1 Macro Scope
2 -----
3 AUTOMATIC
4 GLOBAL
```

Program

This program writes the %put statements into the macro variable named List and executes them.

```
ProcSQL-D-Macros-select-into-list-ordered.sas
1 %Let Z = the last one;
2 %Let A = 1st item;
3 %Let M = middle;
4 Proc SQL; select catx(' ','%Put',Name
5                   ,':' ,Value
6                   ,';'
7                   )
8                   into :List separated by ' '
9                   from Dictionary.Macros
10                  where Scope eq 'GLOBAL'
11                  and not(Name like 'SQL%')
12                  order by Name;
13                  quit;
14 &List.;
```

Notes:

quit : The statements in the macro variable are SAS statements, therefore they must be executed *after* the quit statement.

Exclusion : The clause not (Name like 'SQL%') excludes the macro variables created by the proc sql statement: SqlObs, SqlLoops, SqlXobs, and SqlRc.

List Processing Statements

The statements in macro variable List:

```
ProcSQL-D-Macros-select-into-list-ordered.lst
1 %Put A : 1st item ;
2 %Put M : middle ;
3 %Put Z : the last one ;
```

Program Output

The statements written to the log are a sorted list of global macro variables.

```
ProcSQL-D-Macros-select-into-list-ordered.log
1 14      &List.;
2 A : 1st item
3 M : middle
4 Z : the last one
```

Dictionary Options

Overview

Let's take a look at Dictionary.Options. These next programs show how to find out:

- definitions and values of each option
- what options are in each group

Problem Statement

The task is to write these statements:

```
Proc Options define value option = OptName;
```

Review Data Structure

What is the data structure?

```
_____ ProcSQL-D-Options-describe-table.log _____
1 create table DICTIONARY.OPTIONS
2   (
3     optname char(32) label='Option Name',
4     opttype char(8) label='Option type',
5     setting char(1024) label='Option Setting',
6     optdesc char(160) label='Option Description',
7     level char(8) label='Option Location',
8     group char(32) label='Option Group'
```

Note: group is available in v9.

Program

This program writes text for each option.

```
_____ ProcSQL-D-Options-define-value.sas _____
1 PROC SQL; select catt('Proc Options define value option = '
2                      ,OptName
3                      ,';'
4                      )
5                      into :List separated by ' '
6                      from Dictionary.Options;
7                      quit;
8 &List.;
9 run;
```

List Processing Statements

These statements are in macro variable List:

```
_____ ProcSQL-D-Options-define-value.lst _____
1 Proc Options define value option =APPLETLOC;
2 Proc Options define value option =ARMAGENT;
3 ...
4 Proc Options define value option =XCMD;
```

Program Output

Here is the list of values and definitions for the first option written to the log.

```
1 ProcSQL-D-Options-define-value.log
2 Option Value Information For SAS Option APPLETLOC
3   Option Value: C:\Program Files\SAS\Shared Files\applets\9.1
4   Option Scope: SAS Session
5   How option value set: Config File(s)
6 Option Definition Information for SAS Option APPLETLOC
7   Group= ENVFILES
8   Group Description: SAS library and file location information
9   Description: Location of Java applets
   Type: The option value is of type CHARACTER
```

Options in Each Group

In program `ProcSQL-D-Dictionaries-list-describe-table` above, using the `distinct` function, we created a table of the unique dictionaries before we could write the text of the `describe table` statements. We can reduce two statements to one by creating a named column with the distinct values and then refer to the new column in our text concatenation. This trick requires that we make two macro variables, `Item` and `List`, for each of the columns: `Group` as `Item`, and `text`.

Using Calculated Columns

This example shows that the variable value can be used more than once, and there is no limit to how much text can be concatenated, either before or after use of the variable value.

```
1 ProcSQL-D-Options-list-distinct-groups.sas
2 Proc SQL; select distinct Group as Item,
3             catx(' ', '%Put Group: '
4                 , calculated Item
5                 , ';Proc Options group ='
6                 , calculated Item
7                 , ';run;'
8             )
9             into :Item, :List separated by ' '
10            from Dictionary.Options;
11 quit;
&List.;
```

The `calculated` keyword indicates that the column `Item` has been created — `distinct Group as Item` — and is not in the the table being read: `from Dictionary.Options`.

List Processing statements

Note that the column `Item` retains the label of `Group`.

```
1 ProcSQL-D-Options-list-distinct-groups.lst
2 Option Group
3 -----
4 COMMUNICATIONS  %Put Group: COMMUNICATIONS ;
5                 Proc Options group = COMMUNICATIONS ;run;
6 DATAQUALITY   %Put Group: DATAQUALITY ;
7                 Proc Options group = DATAQUALITY ;run;
```

The `run` statement ensures that the the `%Put Group:` statement is written before each group.

Program Output

This is the first group written to the log.

```
1 12          &List.;
2 Group: COMMUNICATIONS
3     SAS (r) Proprietary Software Release 9.1  TS1M3
4
5     NOAUTOSIGNON    SAS/CONNECT remote submit will not
6                     automatically attempt to SIGNON
7     COMAMID=TCP     Specifies the communication access method
8                     to be used for SAS distributed products
```

Writing Macro Calls

Overview

Macros provide the ability to generate more than a single statement and also conditional processing of parameter values.

In the following sections we examine program templates for processing the three most commonly used lists:

- columns (variables)
- tables (data set names)
- files

Several macro templates are provided for your later usage that show minimal processing of the parameters passed.

In this section:

Macros	18
Dictionary Columns	20
Dictionary Tables	23
Filenames	25

Macros

Macro ProcDsn

This macro is an example program that processes a data set in a library.

```
ProcDsn.sas
1  /*    name: ProcDsn.sas
2  description: process a data set
3  purpose    : called by SQL list processing
4  usage:
5  %ProcDsn(Data=SAShelp.Prdsal2,MemType=data);
6  *****/
7  %Macro ProcDsn(Data    = &Libname..&MemName. /* global */
8              ,MemType = data                /* catalog view */
9              ,Testing = 0);
10
11  *upcase for logical expression evaluation;
12  %let MemType= %upcase(&MemType.);
13
14  %let Testing = %eval(&Testing
15                  or   %sysfunc(getoption(mprint)) eq MPRINT);
16
17  %if &Testing %then %put _local_;
18
19  %if          &MemType. eq CATALOG %then %do;
20      Proc Catalog catalog = &Data.;
21          contents;
22          quit;
23  %end;
24  %else %if &MemType. eq DATA %then %do;
25      Proc SQL; describe table &Data.;
26          quit;
27  %end;
28  %else %if &MemType. eq VIEW %then %do;
29      Proc Contents data = &Data.;
30      run;
31  %end;
32  %else %Put &Data. type unknown: &type.;
33
34  title3 "&Data. type: &MemType.";
35  %Mend;
```

Macro ProcFile

This macro is an example program that processes a file in a folder.

```
ProcFile.sas
1  /*    name: ProcFile.sas
2  description: process a file in a folder
3  purpose    : called by SQL list processing
4  *****/
5  %Macro ProcFile(Directory = &Folder. /*global*/
6              ,Filename = );
7  %if %index(&Filename,..) %then
8      %put note: filename.ext:<&FileName.>;
9  %else %put note: other<&FileName.>;
10  run; %mend;
11
```

Macro ProcVar

This macro is an example program that processes character and numeric variables with different procedures.

```
1  /*      name: ProcVar.sas
2  description: process a variable
3  purpose    : called by SQL list processing
4  usage:
5  %ProcVar(Data=SAShelp.Class,Name=Sex,Type=char)
6  NOTE: SQL dictionary.tables.type in (char,num)
7        compare Contents.type in (1=num,2=char)
8  *****/
9  %Macro ProcVar(Data      = &Libname..&MemName. /* global */
10                ,Name     = Sex
11                ,Type      = char
12                ,Testing   = 0);
13
14  *lowercase for logical expression evaluation;
15  %let Name = %lowercase(&Name.);
16
17  %Let Testing = %eval(&Testing
18                    or %sysfunc(getoption(mprint)) eq MPRINT);
19
20  %If &Testing. %then %put _local_;
21
22  %if          &Type. eq char %then %do;
23      Proc Freq      data    = &Data.;
24                  tables  &Name.;
25  %end;
26  %else %if &Type. eq num %then %do;
27      Proc Summary   data = &Data. print;
28      *Proc Univariate data = &Data.;
29                  var    &Name.;
30  %end;
31
32  title2 "&Data..&Name. type: &Type.";
33  run;%Mend;
```

Dictionary Columns

Overview

Dictionary Columns contains the attributes of each variable in a data set, including name, type, length, label and format.

Problem Statement

The task is to process every variable in a data set, or: every column in a table.

Review Data Structure

```
_____ ProcSQL-D-Columns-describe-table.sas _____  
1 PROC SQL; describe table Dictionary.Columns;  
2 quit;
```

```
_____ ProcSQL-D-Columns-describe-table.log _____  
1 create table DICTIONARY.COLUMNNS  
2 (  
3     libname char(8) label='Library Name',  
4     memname char(32) label='Member Name',  
5     memtype char(8) label='Member Type',  
6     name char(32) label='Column Name',  
7     type char(4) label='Column Type',
```

Program

This program processes every variable in a data set, using the macro ProcVar.

```
_____ ProcSQL-D-Columns-select-into-list.sas _____  
1 options mprint ; *testing: view macro statements;  
2 options source2; *testing: echo include to log;  
3  
4 %Let SQLprint = print; *testing;  
5 *Let SQLprint = noprint;  
6  
7 %Let LibName = SASHELP;  
8  
9 %Let MemName = Class;  
10 *Let MemName = PrdSal2;  
11 *Let MemName = PrdSal3;  
12 *Let MemName = PrdSale;  
13  
14 %Let MemType = data;  
15 *Let MemType = view;  
16  
17 %Include 'ProcVar.sas';  
18  
19 PROC SQL &SQLprint.;  
20     select catt('%ProcVar(name=' , Name  
21                 ',type=' , Type  
22                 ',')'  
23                 )  
24     into :List separated by ' '  
25     from Dictionary.Columns  
26     where LibName eq "%upcase(&LibName.)"  
27           and MemName eq "%upcase(&MemName.)"  
28           and MemType eq "%upcase(&MemType.)";  
29     quit;  
30 &List.; *execute;
```

Notes:

Testing : Place an asterisk (*) before the options statements to disable either or both of the macro option mprint or include option source2.

Change the %Let verb to *Let to disable SQL print of query results.

Parameters : The program has three required parameters:

LibName : The primary parameter is the libref LibName.

MemName : The secondary parameter is the table name, MemName, in the libref.

MemType : A tertiary parameter, provided for clarity, is the member type, MemType, in (data, view).

Macro : The %Include statement brings the macro definition into the program.

List Processing Statements

These are the statements in the macro variable List. Spaces have been added for clarity.

```
ProcSQL-D-Columns-select-into-list.lst
1 %ProcVar(name=Name ,type =char)
2 %ProcVar(name=Sex ,type =char)
3 %ProcVar(name=Age ,type =num )
4 %ProcVar(name=Height,type =num )
5 %ProcVar(name=Weight,type =num )
```

Program Output

The macro ProcVar generates steps.

```
ProcSQL-D-Columns-select-into-list.lst
1 58 &List.; *execute;
2 MPRINT(PROCVAR): Proc Freq data = SAShelp.Class;
3 MPRINT(PROCVAR): tables Name;
4 ...
5 MPRINT(PROCVAR): Proc Freq data = SAShelp.Class;
6 MPRINT(PROCVAR): tables Sex;
7 ...
8 MPRINT(PROCVAR): Proc Summary data = SAShelp.Class print;
9 MPRINT(PROCVAR): var Age;
```

Problems

There are limits associated with this method.

catt : The catt function returns a value whose maximum length is 200 in proc sql.

```
_____ RnD-catt-function-warning.log _____
1  WARNING: In a call to the CATT function, the buffer allocated
2      for the result was not long enough to contain the
3      concatenation of all the arguments. The correct
4      result would contain 200+ characters, but the actual
5      result may either be truncated to 200 character(s) or
6      be completely blank, depending on the calling
7      environment.
```

macro variable : The maximum length of a macro variable is $2^{16} - 2 = 65534$.

These statements show the error.

```
_____ RnD-macro-variable-length-error.log _____
1  2      %let MaxLen = %eval(2**16);
2  3      %put MaxLen = &MaxLen.;
3  MaxLen = 65536
4  4      %let List = %sysfunc(repeat(*,&MaxLen.-2));
5  ERROR: The text expression length (65535)
6      exceeds maximum length (65534). The text expression
7      has been truncated to 65534 characters.
```

Workarounds

These are some workarounds for consideration:

- replace catt function with concatenation operator: double bang or exclamation points: (!!) or double vertical bars (||)
- reduce macro name to one character
- replace named parameters with positional parameters
- eliminate macro parameters when information can be discovered by the subroutine macro; e.g.: variable name is a foreign key whose associated information — type, etc. — can be gotten by subroutine macro ProcVar.

Here is an example program showing the workarounds.

```
1  %let LibName = SASHELP;
2  %let MemName = CLASS;
3  *previous: Macro ProcVar (data=,name=,...);
4  %Macro V(Name      /* positional parameter */
5              /* global macro vars: */
6              ,Data   = &LibName..&MemName.
7              ,Testing = 0);
8  *todo: get information: type, format, etc.;
9  ...
10 PROC SQL &SQLPrint.;
11     select '%V(' !! trim(Name) !! ' )'
12     ...
```

Dictionary Tables

Overview

This program illustrates the concept of having one program define the parameters of another.

The task is to process every data set in a library. What are the columns whose values are to be passed to the processing macro?

Review Data Structure

This program lists the data structure of Dictionary Tables;

```
1 _____ ProcSQL-D-Tables-describe-table.sas _____
2 PROC SQL; describe table Dictionary.Tables
3   ; quit;
4 run;

1 _____ ProcSQL-D-Tables-describe-table.log _____
2 create table DICTIONARY.TABLES
3   (
4     libname char(8) label='Library Name',
5     memname char(32) label='Member Name',
6     memtype char(8) label='Member Type',
7     nobs num label='Number of Physical Observations',
8     ...
9     nvar num label='Number of Variables',
```

Program

This program processes every data set in a library, using the macro ProcDsn.

```
1 _____ ProcSQL-D-Tables-select-into-list.sas _____
2 options mprint;*testing: view macro statements;
3 options source2;*testing: echo include to log;
4
5 %Let SQLprint = noprint;*production;
6 %Let SQLprint = print;*testing;
7
8 %Let LibName = SASHELP;*testing;
9
10 * Libname MyLib '<folder-name>';
11 *Let LibName = MyLib; *production;
12
13 %Let MemType = 'data'; *NOTE: MUST BE QUOTED;
14 *Let MemType = 'catalog' 'data' 'view';
15
16 %Include 'ProcDsn.sas';
17
18 %Let List = *empty: no rows selected;
19
20 PROC SQL &SQLprint.;
21   select catt('%ProcDsn(data=',MemName
22     ,',type ='      ,MemType
23     ,',nobs ='      ,Nobs
24     ,',nvar ='      ,Nvar
25     ,')'
26   into :List separated by ' '
27   from Dictionary.Tables
28   where LibName eq "%upcase(&LibName.)"
29     and MemType in (%upcase(&MemType.));
30   quit;
```

```

31 &List. ;*execute macro calls;
32 %symdel List;

```

Testing : Place an asterisk (*) before the options statements to disable either or both of the macro option mprint or include option source2.

Change the %Let verb to *Let to disable SQL print of query results.

Parameters : These parameters are required.

LibName : The primary parameter is the libref, LibName.

MemType : Libraries contain catalogs, data sets, and views. Remember to provide the value in single or double quotes.

List Processing Statements

As in the processing of Dictionary.Columns program above, the statements written include most of the list attributes to be passed to the macro.

```

_____ ProcSQL-D-Tables-select-into-list.lst _____
1 %ProcDsn(data=ADOMSG,type =DATA,nobs = 458,nvar =6)
2 %ProcDsn(data=ADSMSG,type =DATA,nobs = 426,nvar =6)
3 %ProcDsn(data=AFMSG ,type =DATA,nobs = 1088,nvar =6)
4 ...
5 %ProcDsn(data=ZTC ,type =DATA,nobs =18161,nvar =6)

```

Program Output

```

_____ ProcSQL-D-Tables-select-into-list.lst _____
1 65 &List. ;*execute macro calls;
2 SAShelp.ADOMSG nobs: 458 nvar: 6
3 MPRINT(PROCDN): Proc SQL;
4 MPRINT(PROCDN): describe table SAShelp.ADOMSG;
5 NOTE: SQL table SASHELP.ADOMSG was created like:

```

Compare with program ProcSQL-D-Columns-select-into-list, above.

FileNames

Overview

Processing a list of filenames is different from the previous examples as there is no SQL dictionary of folders and filenames. Reading the list of files using SQL requires some interesting tricks. This example is a polished version of a program written by Hamilton [25, sugi31.046]

Program

This program processes every file in a folder, using the macro ProcFile. It uses the scl functions: filename, dopen, and dread.

```
1      ProcSQL-filenames-select-into-list.sas
2      /*      Name: ProcSQL-select-into-list-from-filenames.sas
3
4      Requirements : description : Process Each File in Folder
5                   purpose      : list processing: read file list
6                               process each file
7
8      -----
9      Contexts      : program group: list processing
10                   program type: routine
11                   SAS type: program with macro
12                   uses routines: in-program macro %ProcFile
13
14      -----
15      Specifications: input : folder: directory-specification
16                      process: open folder
17                          read filenames
18                          SQL writes macro calls of ProcFile
19                          execute list: ProcFile of each filename
20                      output : user-defined in macro ProcFile
21
22      -----
23      Usage Examples:
24      %Let Folder = c:\;
25
26      Information : concept author: Jack Hamilton,
27                   DIGITS and DATES The SQL Procedure Goes "Loopy"
28                   Proceedings of the 31st Annual SAS(R) Users
29                   Group International Conference, 2006.
30                   http://www2.sas.com/proceedings/sugi31/046-31.pdf
31                   polishing: Ronald J. Fehd
32                   for PNWSUG-2006 Hands On Workshop
33
34      NOTE: The execution of this query involves performing one
35            or more Cartesian product joins that can not be optimized.
36      *** ..... */
37
38      %Let Folder = c:\;
39      *Let Folder = .; *here;
40      *Let Folder = %sysget(sasroot); *directory-spec;
41      *Let Folder = %sysget(sasroot)\core\sasexe; *922 files;
42
43      options mprint ;*testing: echo macro statements;
44      options source2;*testing: echo included statements;
45
46      %Let SQLprint = noprint; *production;
47      %Let SQLprint = print; *testing;
48
49      %Include 'ProcFile.sas';
50
51      Data Work.Digits;
52      length Digit 4;
53      do Digit = 0 to 9;
54          output;
55      end;
56      stop;
57      run;
58
59      * see: from (&&E&Evalue.);
60      %Let E1 = select ones.digit as FileNmbr from Digits
```

```

59                                     as ones;
60 %Let E2 = select ones.digit + 10 * tens.digit
61                                     as FileNmbr
62                                     from Digits as ones, digits as tens;
63 %Let E3 = select ones.digit + 10 * tens.digit
64                                     + 100 * hundreds.digit
65                                     as FileNmbr
66                                     from Digits as ones, digits as tens,
67                                     digits as hundreds;
68 %Let E4 = select ones.digit + 10 * tens.digit
69                                     + 100 * hundreds.digit
70                                     + 1000 * thousands.digit
71                                     as FileNmbr
72                                     from Digits as ones, digits as tens,
73                                     digits as hundreds,
74                                     digits as thousands;
75 Proc SQL &SQLprint.;
76     select filename('DirSpec', "&Folder."),
77            dopen('DirSpec') as Dir_id,
78            dnum(calculated Dir_id)
79            into :FileName_Rc, :Dir_id, :NmbrFiles
80            from Digits where Digit = 0;
81 %Let NmbrFiles = &NmbrFiles.; *trim;
82 %Let Evalvalue = %length(&NmbrFiles.);
83 select catt('%ProcFile(filename='
84            ,dread(&Dir_id., FileNmbr)
85            ,')'
86            )
87            into :List separated by ' '
88            from (&E&Evalvalue.)
89            where FileNmbr between 1 and &NmbrFiles;
90 quit;
91 %Let FileName_Rc = dclose(&Dir_id.);
92 &List.;
93 %symdel E1 E2 E3 E4 Evalvalue
94 Folder List SQLprint
95 FileName_Rc Dir_id NmbrFiles;

```

Testing : Place an asterisk (*) before the options statements to disable either or both of the macro option mprint or include option source2.

Change the %Let verb to *Let to disable SQL print of query results.

Parameters :

Folder : The primary parameter is the folder name.

Macro : The %Include statement brings the macro definition into the program.

Data Set : The Digits data set is used to make a larger data set, FileNmbrs, containing the file-numbers used by the dread function.

Macro Variables E* : Macro variables E1–E4 are necessary for processing lists of 9, 99, 999 or 9999 files.

Select : Two select statments are required: the first to assign a fileref, open the folder and read the number of files present; the second statement writes the macro calls.

Using Macro %Do Loops

Overview

This section provides two methods which replace the `select ...into :List` clause which is fragile.

These methods are:

- delimited list: one macro variable contains the list of values and each value is delimited by a special character
- macro array: a sequentially-numbered set of macro variables

These methods are better than the set of workarounds shown above.

Note: Programs shown here are calling routines for macro ProcDsn. The zip file provided contains similar programs for macro ProcVar:

- ProcVars
- ProcVars-caller
- ProcVars-Test
- ProcVarsDL
- ProcVarsDL-Test

List processing techniques shown here are from Fehd and Carpenter [22, sgf2007.113].

In this section:

Using Items in Delimited List	28
Using Arrays of Macro Variables	30

Using Items in Delimited List

Overview

This method stores all values in one macro variable with each item separated by a delimiter; the items are retrieved for processing using the scan function.

```
1 demo-delimited-list.log
2 1      %let List = ItemA * ItemB * itemC;
3 2      %let Item1 = %scan(&List.,1,*);
4 3      %put Item1: &Item1.;
5 Item1: ItemA
6 4      %let Item3 = %scan(&List.,3,*);
7 5      %put Item3: &Item3.;
8 Item3: itemC
```

Program

This program processes every data set in a libref, using the macro ProcDsn and the method of one macro variable containing a delimited list of all values.

```
1 ProcDsnSDL.sas
2 /* name: ProcDsnSDL.sas
3 description: process all data sets in libref
4 using named macro
5 purpose : list processing
6 note : uses delimited list of values
7 usage:
8 %ProcDsn(libref=SAShelp,macroname=ProcDsn);
9 *****/
10 %Macro ProcDsnDL(LibName = SAShelp
11 ,MemType = data /* or view */
12 ,MacroName = put ProcDsn
13 ,SQLprint = noprint
14 ,Testing = 0);
15
16 %let Testing = %eval(&Testing
17 or %sysfunc(getoption(mprint)) eq MPRINT);
18
19 %if &Testing. %then %let SQLprint = print;
20
21 %local Dlm I;
22 %let Dlm = *;
23
24 PROC SQL &SQLprint.;
25 select MemName, MemType
26 into :List_Name separated by "&Dlm."
27 , :List_Type separated by "&Dlm."
28 from Dictionary.Tables
29 where LibName eq "%upcase(&LibName.)"
30 and MemType eq "%upcase(&MemType.)";
31 quit;
32
33 %if &Testing %then %put _local_;
34
35 %do I = 1 %to &SqlObs.;
36 %let Item_Name = %scan(&List_Name,&I.,&Dlm.);
37 %let Item_Type = %scan(&List_Type,&I.,&Dlm.);
38 %if &Testing %then %do;
39 %put note2:Item_Name&I: &Item_Name.;
40 %put note2:Item_Type&I: &Item_Type.;
41 %end;
42 %&MacroName.(libname=&Item_Name.,memtype=&Item_Type.);
43 %end;
44 run;
45 %mend;
```

Program: Testing

This program provides a unit test of the routine ProcDsnsDL.

```
----- ProcDsnsDL-Test.sas -----
1  *name: ProcDsnsDL-Test;
2  options mprint ; *testing: view macro statements;
3  *either of:
4  *autoexec;
5  *filename Project '.';
6  *options sasautos = (Project SASautos);
7  *
8  *OR;
9  %Include 'ProcDsnsDL.sas' 'ProcDsn.sas';
10 *unit test;
11 *default = SAShelp;
12 %ProcDsns();
13
14 %ProcDsnsDL(LibName=SAShelp,MacroName = ProcDsn);
15
16 *Libname Library '<directory-specification>';
17 *ProcDsnsDL(LibName=Library);
```

Using Arrays of Macro Variables

Overview

This method stores each value in a macro variable; the items are retrieved for processing using an index variable.

```
demo-macro-array.log
1      %Macro DemoMacroArray();
2      %let Item1 = valueA;
3      %let Item2 = value.B;
4      %let Item3 = value C;
5      %let Dim = 3;
6      %do I = 1 %to &Dim.;
7          %Put Item&I.: &&Item&I.;
8      %end;
9      %Mend;
10     %DemoMacroArray
11 Item1: valueA5
12 Item2: value.B
13 Item3: value C
14
```

Program

This program processes every data set in a libref, using the macro ProcDsn and the macro array method of allocating a series of sequentially-numbered macro variables for each value.

```
ProcDsns.sas
1      /* name: ProcDsns.sas
2      description: process all data sets in libref
3                  using named macro
4      purpose      : list processing
5      usage:
6      %ProcDsns(libref=SAShelp,macroname=ProcDsn);
7      *****/
8      %Macro ProcDsns(LibName = SAShelp
9                      ,MemType = data /* or view */
10                     ,MacroName = put ProcDsn
11                     ,SQLprint = noprint
12                     ,Testing = 0);
13
14      %local I;
15      %let I = 0;
16
17      %let Testing = %eval(&Testing
18                          or %sysfunc(getoption(mprint)) eq MPRINT);
19
20      %if &Testing. %then %let SQLprint = print;
21
22      PROC SQL &SQLprint.;
23          select MemName, MemType
24          into :MemName1 - :MemName&SysMaxLong.
25              , :MemType1 - :MemType&SysMaxLong.
26          from Dictionary.Tables
27          where LibName eq "%upcase(&LibName.)"
28                and MemType eq "%upcase(&MemType.)";
29          quit;
30
31      %if &Testing %then %put _local_;
32
33      %do I = 1 %to &SqlObs.;
34          %MacroName.(data=&Libname..&&MemName&I., memtype=&&MemType&I.);
35      %end;
36      run;
37      %Mend;
```

Notes on program ProcDsns:

- Testing : Fehd [19, nesug07.cc12] shows the discipline of using options to turn on self-reporting by macro routines.
- SysMaxLong : This macro variable contains the largest integer available to the operating system and is used to ensure the upper bound of the macro array is not truncated.
- Usage here replaces any smaller number — :MemName9999 — which may result in allocating less than the number of macro variables of the rows in the table; i.e.: no macro variable MemName10000 is created.
- Double dots : The prefix of a macro variable reference is an ampersand (&); the suffix of a reference is a dot (.). The dot suffix is necessary for a two-level data set name: Work.Data:
- &Libname..&MemName.
- Item reference : A reference to an item in the macro array contains the array name and the index. The array name must be preceeded by two ampersands in order to resolve correctly:
- &&MemName.&I. :: &MemName1

Program: Module

This program is a module and provides a template for use of the macro routine ProcDsns.

```
ProcDsns-caller.sas
1  *name: ProcDsns-caller;
2  options mprint ; *testing: view macro statements;
3  *either of:
4  *autoexec;
5  *filename Project '.';
6  *options sasautos = (Project SASautos);
7  *
8  *or;
9  %Include 'ProcDsns.sas' 'ProcDsn.sas';
10 %ProcDsns(Libname=SAShelp,MacroName = ProcDsn);
11
12 *Libname Library '<directory-specification>';
13 *ProcDsns(Libname=Library);
```

Conclusion

Any data set is a candidate for use by list processing. To produce dynamic programs follow these simple steps:

- identify the data set (table)
- examine its data structure
- identify the variables (columns) that contain parameter values
- develop a program with example code
- use proc sql to write that code as text or macro call, substituting variable names for values
- sit back and watch the log zoom by

Summary : be aware of limitations:

- maximum length of 200 for cat functions
- maximum length of one macro variable: $2^{16} - 2 = 65534$

Recommendation : Use select into macro arrays, they are the easiest to debug, test and understand.

Acknowledgements

Ian Whitlock and Sig Hermansen have piqued my interest in SQL over the years with their contributions to SAS-L. Toby Dunn provided commentary on numerous examples. Jack Hamilton provided the file-handling program.

Suggested Readings

- bookshelf** : Carpenter [6, saspress.59224], Celko [9, Celko.2000], Celko [10, Celko.2005] Schreier [35, saspress.60500]
- basics** : Dickstein and Pass [13, sugi29.269], Hu [27, sugi29.042], Lund [31, sugi30.257], Ronk [34, sugi29.268], Wells [39, sugi26.105], Winn, Jr. [43, sugi22.067]
- intermediate** : and advanced concepts: Barber [4, sugi22.198], Hamilton [25, sugi31.046] Hermansen [26, sugi22.035] Lafler [28, sugi28.019], Loren and Nelson [30, sugi23.031], Winn, Jr. [44, sugi23.035]
- list processing** : Abolafia [1, sugi30.031], Andrews [3, sugi31.039], Beakley and McCoy [5, sugi29.078], ch. 9, dynamic programming, Carpenter [6, saspress.59224], Carpenter [7, sugi30.028], Fehd [20, sgf2007.028], Fehd and Carpenter [22, sgf2007.113], Fehd [21, sgf2008.003], Pollock [33, sugi30.057], Varney [38, sugi31.045]
- macros and SQL** : Adams [2, sugi28.087], Chiu and Heaton [11, sugi28.097], Delaney and Carpenter [12, sugi29.128], Dilorio and Abolafia [15, sugi29.237], Dilorio [14, sugi30.268], Droogendyk and Fecht [16, sugi31.251], Fehd [18, sugi29.070], First and Ronk [24, sugi31.107], Stroupe [36, sugi28.056], Sun and Wong [37, sugi30.040], Whitlock [41, sugi29.244], Whitlock [42, sugi30.252]
-

Bibliography

- [1] Jeff Abolafia. What would I do without proc SQL and the macro language? In *Proceedings of the 30th SAS User Group International Conference*, 2005. URL <http://www2.sas.com/proceedings/sugi30/031-30.pdf>. Coders' Corner, 5 pp.; creating and using arrays of macro variables, creating partial sas statements using catx function.
- [2] John H. Adams. The power of recursive sas macros — how can a simple macro do so much? In *Proceedings of the 28th SAS User Group International Conference*, 2003. URL <http://www2.sas.com/proceedings/sugi28/087-28.pdf>. Coders' Corner, 5 pp.; reading directories and subdirectories using recursive macro calls.
- [3] Rick Andrews. Sas macro dynamics — from simple basics to powerful invocations. In *Proceedings of the 31st SAS User Group International Conference*, 2006. URL <http://www2.sas.com/proceedings/sugi28/087-28.pdf>. Coders' Corner, 6 pp.; creating macro variables, comparison of SQL and call symput, making and using lists of values in macro variables, writing to file then including, using call execute for list processing, creating and using array of macro variables, bibliography.
- [4] Brenda M. Barber. Overcoming kainophobia: Replacing complex merges with proc sql. In *Proceedings of the 22nd SAS User Group International Conference*, 1997. URL <http://www2.sas.com/proceedings/sugi22/POSTERS/PAPER198.PDF>. Poster, 6 pp.; merging using SQL joins.
- [5] Steven Beakley and Suzanne McCoy. Dynamic sas programming techniques, or how not to create job security. In *Proceedings of the 29th SAS User Group International Conference*, 2004. URL <http://www2.sas.com/proceedings/sugi29/078-29.pdf>. Coders' Corner, 10 pp.; using SQL to make lists of values, example programs.
- [6] Arthur L. Carpenter. *Carpenter's Complete Guide to the SAS Macro Language, Second Edition*. Cary, NC: SAS Institute Inc., 2004. URL <http://www.sas.com/apps/pubscat/bookdetails.jsp?pc=59224>. 13 chap., 475 pp., appendices: 5, glossary: 3 pp., bibliography: 19 pp., index: 13 pp.
- [7] Arthur L. Carpenter. Storing and using a list of values in a macro variable. In *Proceedings of the 30th SAS User Group International Conference*, 2005. URL <http://www2.sas.com/proceedings/sugi30/028-30.pdf>. Coders' Corner, 6 pp.; making macro arrays using symputx or sql, iterating macro arrays, using scl functions, bibliography.
- [8] Arthur L. Carpenter. The path, the whole path, and nothing but the path, so help me windows. In *Proceedings of the SAS Global Forum*, 2008. URL <http://www2.sas.com/proceedings/forum2008/023-2008.pdf>. Applications Development, 8 pp.; pathname function, dictionary.libnames, how to get a UNC.
- [9] Joe Celko. *Joe Celko's SQL for Smarties: Advanced SQL Programming, Second Edition*. Morgan-Kaufmann, San Francisco, CA, USA, 2000. URL <http://www.celko.com>.
- [10] Joe Celko. *Joe Celko's SQL Programming Style*. Elsevier, San Francisco, CA, USA, 2005. URL <http://www.elsevier.com>. 10 chap., 216 pp., bibliography: 3 pp., index: 16 pp.
- [11] Grace Chiu and Edward Heaton. An efficient approach to combine SAS data sets with voluminous variables that need name and other changes. In *Proceedings of the 28th SAS User Group International Conference*, 2002. URL <http://www2.sas.com/proceedings/sugi28/097-28.pdf>. Coders' Corner, 5 pp.; renaming variables, proc contents and proc sql.
- [12] Kevin P. Delaney and Arthur L. Carpenter. SAS macro: Symbols of frustration? %Let us help! a guide to debugging macros. In *Proceedings of the 29th SAS User Group International Conference*, 2002. URL <http://www2.sas.com/proceedings/sugi29/128-29.pdf>. Hands-on Workshops, 20

pp.; using options to aid debugging, scan and qscan functions, problems when using single or double quotes, SQL selecting values into macro variable.

- [13] Craig Dickstein and Ray Pass. Data step vs. proc sql: What's a neophyte to do? In *Proceedings of the 26th SAS User Group International Conference*, 2001. URL <http://www2.sas.com/proceedings/sugi29/269-29.pdf>. Beginning Tutorials, 9 pp.; comparison of merge and joins, using proc SQL for recoding, SQL functions.
- [14] Frank Dilorio. Rules for tools — the SAS utility primer. In *Proceedings of the 30th SAS User Group International Conference*, 2005. URL <http://www2.sas.com/proceedings/sugi30/268-30.pdf>. Tutorials, 19 pp.; reusing macros and programs, sql dictionary tables.
- [15] Frank Dilorio and Jeff Abolafia. Dictionary tables and views: Essential tools for serious applications. In *Proceedings of the 29th SAS User Group International Conference*, 2002. URL <http://www2.sas.com/proceedings/sugi29/237-29.pdf>. Tutorials, 19 pp.; comparison of data structure of dictionary tables and sashelp views, review of differences between v6, v8, and v9, 12 examples.
- [16] Harry Droogendyk and Marje Fecht. Demystifying the SAS macro facility — by example. In *Proceedings of the 31st SAS User Group International Conference*, 2006. URL <http://www2.sas.com/proceedings/sugi31/251-31.pdf>. tutorials, 15 pp.; list processing, macro functions: symdel, symexist, syslput, sysrput; proc sql, sql functions: open, attrn, close.
- [17] Peter Eberhardt and Ilene Brill. How do i look it up if I cannot spell it: An introduction to SAS(R) dictionary tables. In *Proceedings of the SAS Global Forum*, 2007. URL <http://www2.sas.com/proceedings/forum2007/235-2007.pdf>. Tutorials, 14 pp.; reviews many sql dictionaries, example list processing programs, bibliography.
- [18] Ronald Fehd. Array: Construction and usage of arrays of macro variables. In *Proceedings of the 29th SAS User Group International Conference*, 2002. URL <http://www2.sas.com/proceedings/sugi29/070-29.pdf>. Coders' Corner, 6 pp.; using proc sql, bibliography.
- [19] Ronald J. Fehd. Writing testing-aware programs that self-report when testing options are true. In *Proceedings of the NorthEast SAS User Group Annual Conference*, 2007. URL <http://www.nesug.org/proceedings/nesug07/cc/cc12.pdf>. Coders' Corner, 20 pp.; use of options echoauto, mprint, source2, and verbose during unit and integration testing, bibliography.
- [20] Ronald J. Fehd. Journeymen's tools: Data review macro FreqAll – using Proc SQL list processing with Dictionary.Columns to eliminate macro do loops. In *Proceedings of the SAS Global Forum*, 2007. URL <http://www2.sas.com/proceedings/forum2007/028-2007.pdf>. Coder's Corner, 10 pp.; attributes, dictionary.columns, metadata, proc append, proc freq, proc sql, program header; bibliography.
- [21] Ronald J. Fehd. SmryEachVar: a data-review routine for all data sets in a libref. In *Proceedings of the SAS Global Forum*, 2008. URL <http://www2.sas.com/proceedings/forum2008/003-2008.pdf>. Applications Development, 24 pp.; successor to FreqAll, list processing using parameterized includes, utilities which write: 1. attribute statements 2. format statements, used to provide missing variable attributes, bibliography.
- [22] Ronald J. Fehd and Art Carpenter. List processing basics: Creating and using lists of macro variables. In *Proceedings of the SAS Global Forum*, 2007. URL <http://www2.sas.com/proceedings/forum2007/113-2007.pdf>. Hands On Workshop, 20 pp.; comparison of methods: making and iterating macro arrays, scanning macro variable, writing calls to macro variable, write to file then include, call execute; 11 examples, bibliography.
- [23] Ying Feng. The SQL procedure: When and how to use it. In *Proceedings of the 31st SAS User Group International Conference*, 2006. URL <http://www2.sas.com/proceedings/sugi31/044-31.pdf>. Coders' Corner, 7 pp.; using SQL to make and iterate a macro array, bibliography.

- [24] Steven First and Katie Ronk. Intermediate and advanced SAS macros. In *Proceedings of the 31st SAS User Group International Conference*, 2006. URL <http://www2.sas.com/proceedings/sugi31/107-31.pdf>. Hands-on Workshops, 16 pp.; call routines: execute, symdel, symput, symputn, symputx; functions: resolve, symexist, symget, symgetn, symglobal, symlocal, sysexec, sysget; macro arrays, proc sql, sas/connect: syslput, sysrput.
- [25] Jack Hamilton. Digits and dates: The SQL procedure goes loopy. In *Proceedings of the 31st SAS User Group International Conference*, 2006. URL <http://www2.sas.com/proceedings/sugi26/061-26.pdf>. Coders' Corner, 11 pp.; simulating looping thru dates, using scl functions in sql, using date functions in sql.
- [26] Sigurd Hermansen. Ten good reasons to learn SAS software's SQL procedure. In *Proceedings of the 22nd SAS User Group International Conference*, 1997. URL <http://www2.sas.com/proceedings/sugi26/061-26.pdf>. Advanced Tutorials, 5 pp.; comparison of SQL to data step.
- [27] Weiming Hu. Top ten reasons to use proc sql. In *Proceedings of the 29th SAS User Group International Conference*, 2004. URL <http://www2.sas.com/proceedings/sugi29/042-29.pdf>. Coders' Corner, 6 pp.; joining, selecting values into list, text wrapping, summary functions, matching, inserting, using coalesce function, summarizing, fuzzy merges, examples, bibliography.
- [28] Kirk Paul Lafler. Undocumented and hard-to-find SQL features. In *Proceedings of the 28th SAS User Group International Conference*, 2003. URL <http://www2.sas.com/proceedings/sugi28/019-28.pdf>. Advanced Tutorials, 6 pp.; case logic, coalesce function, SQL statement options _method and _tree, SQL dictionary tables, automatic macro variables, performance issues, examples.
- [29] Stuart Long and Ed Heaton. Using the SAS(R) data step and proc SQL to create macro arrays. In *Proceedings of the SAS Global Forum*, 2008. URL <http://www2.sas.com/proceedings/forum2008/105-2008.pdf>. Applications Development, 11 pp.; comparison of macro arrays and delimited list, symputx.
- [30] Judy Loren and Gregory S. Barnes Nelson. Sql step by step: An advanced tutorial for business users. In *Proceedings of the 23rd SAS User Group International Conference*, 1998. URL <http://www2.sas.com/proceedings/sugi23/Advttutor/p31.pdf>. Advanced Tutorial, 10 pp.; examples, bibliography.
- [31] Pete Lund. An introduction to SQL in SAS. In *Proceedings of the 30th SAS User Group International Conference*, 2005. URL <http://www2.sas.com/proceedings/sugi30/257-30.pdf>. Tutorials, 22 pp.; basic syntax, where clause operators: null, missing, between, contains, like, sounds like; aggregating functions.
- [32] William C. Murphy. Changing data set variables into macro variables. In *Proceedings of the SAS Global Forum*, 2007. URL <http://www2.sas.com/proceedings/forum2007/050-2007.pdf>. Coders Corner, 4 pp.; code reuse, macro arrays and do loops, call set.
- [33] Stuart Pollack. Techniques for effectively selecting groups of variables. In *Proceedings of the 30th SAS User Group International Conference*, 2005. URL <http://www2.sas.com/proceedings/sugi30/057-30.pdf>. Coders' Corner, 4 pp.;
- [34] Katie Minten Ronk. Introduction to PROC SQL. In *Proceedings of the 29th SAS User Group International Conference*, 2004. URL <http://www2.sas.com/proceedings/sugi29/268-29.pdf>. Tutorials, 13 pp.; examples.
- [35] Howard Schreier. *PROC SQL by Example: Using SQL within SAS*. Cary, NC: SAS Institute Inc., 2008. URL <http://www.sas.com/apps/pubscat/bookdetails.jsp?pc=60500>. 14 chap., 275 pp.
- [36] Jane Stroupe. Nine steps to get started using SAS macros. In *Proceedings of the 28th SAS User Group International Conference*, 2003. URL <http://www2.sas.com/proceedings/sugi28/>

056-28.pdf. Beginning Tutorials, 5 pp.; creating macro variables using proc sql, developing macros from programs.

- [37] Helen Sun and Cindy Wong. A macro for importing multiple excel worksheets into SAS data sets. In *Proceedings of the 30th SAS User Group International Conference*, 2005. URL <http://www2.sas.com/proceedings/sugi30/040-30.pdf>. Coders' Corner, 9 pp.; application: import .xls, using scan function in macro do loops, bibliography.
- [38] Brian Varney. Using metadata and project data for data-driven programming. In *Proceedings of the 31st SAS User Group International Conference*, 2006. URL <http://www2.sas.com/proceedings/sugi31/045-31.pdf>. Coders' Corner, 10 pp.; assertions, call execute, call symput, macro arrays, scl functions, SQL dictionary tables, stored processes, subsetting, write to file and %include.
- [39] Clayton Wells. Tips for using proc SQL to extract information from medical claims. In *Proceedings of the 26th SAS User Group International Conference*, 2001. URL <http://www2.sas.com/proceedings/sugi26/p105-26.pdf>. Coder's Corner, 4 pp.; comparison of proc transpose to sql, practical usage, SQL functions.
- [40] Ian Whitlock. Proc sql: Is it a required tool for good sas programming? In *Proceedings of the 22nd SAS User Group International Conference*, 1997. URL <http://www2.sas.com/proceedings/sugi26/p060-26.pdf>. Beginning Tutorials, 6 pp.; using SQL to create macro arrays, selecting distinct text into macro variable.
- [41] Ian Whitlock. A second look at SAS macro design issues. In *Proceedings of the 29th SAS User Group International Conference*, 2004. URL <http://www2.sas.com/proceedings/sugi26/244-29.pdf>. Tutorials, 18 pp.; examples, list processing, macro arrays, macro parameter naming conventions, sashelp views, SQL dictionary tables.
- [42] Ian Whitlock. Macro bugs: How to create, avoid, and destroy them. In *Proceedings of the 30th SAS User Group International Conference*, 2005. URL <http://www2.sas.com/proceedings/sugi30/252-30.pdf>. Tutorials, 20 pp.; call execute, call symput, debugging, design, how macro facility works, options used when testing: mfile mlogic mprint, readability, single and double quoting, scope of macro variables, testing, understanding difference between macro and sas bugs writing messages to log with %put.
- [43] Thomas J. Winn, Jr. Introduction to using proc sql. In *Proceedings of the 22nd SAS User Group International Conference*, 1997. URL <http://www2.sas.com/proceedings/sugi26/p105-26.pdf>. Beginning Tutorial, 7 pp.; create tables and views, examples, joins, operators.
- [44] Thomas J. Winn, Jr. Intermediate proc sql. In *Proceedings of the 23rd SAS User Group International Conference*, 1998. URL <http://www2.sas.com/proceedings/sugi23/Advttutor/p35.pdf>. Advanced Tutorial, 7 pp.; examples, summary functions, views.

Get Programs

To get the code examples in this paper search <http://www.sascommunity.org> for the HOW SQL for List Processing zip.

Author

Author: Ronald J. Fehd
Centers for Disease Control
4770 Buford Hwy NE
Atlanta GA 30341-3724

<mailto:RJF2@cdc.gov>

about the author:

education:	B.S. Computer Science, U/Hawaii,	1986
	SAS User Group conference attendee since	1989
	SAS-L reader	since 1994
experience:	programmer: 20+ years	
	data manager at CDC, using SAS: 17+ years	
	author: 30+ SUG papers	
SAS-L:	author: 5,000+ messages to SAS-L since	1997
	Most Valuable SAS-L contributor:	2001, 2003

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. In the USA and other countries ® indicates USA registration.

Document Production: This paper was typeset in \LaTeX . For further information about using \LaTeX to write your SUG paper, consult the SAS-L archives:

<http://www.listserv.uga.edu/cgi-bin/wa?S1=sas-l>
Search for :
The subject is or contains: \LaTeX
The author's address : RJF2
Since : 01 June 2003

Index

- ampersands, double, 31
- arrays of macro variables, 27, 30
- bang, 7
- calculated columns, using, 16
- calling macros, 17
- catalog procedure, 18
- catt function, 8, 22
- colon, select into, 9
- concatenating
 - functions, 7
 - operator, 7, 22
 - text, 8
- constant text, writing, 8
- contents procedure, 18
- creating macro variables with sql, 8
- dictionary
 - columns, 20
 - dictionaries, 11
 - macros, 13
 - options, 15
 - tables, 23
- distinct, select, 11, 16
- do loops, macro, 27
- dopen function, 25
- double ampersands, 31
- double dots in macro variables, 31
- dread function, 25
- exclamation points, 7
- filename function, 25
- filenames, processing, 25
- fragile, 27
- freq procedure, 19
- functions
 - cat, 7
 - catt, 8, 22
 - concatenation, 7
 - distinct, 11, 16
 - dopen, 25
 - dread, 25
 - filename, 25
 - scan, 28
 - scl, 25
- global macro variables, 13
- into macro variable, 9

- like operator, 9, 14
- list processing summary, 10
- lists, making, 6, 11
- macro arrays, 27, 30
 - upper bound of, 31
- macro do loops, 27
- macro templates, 17
- macro variables
 - creating, with sql, 8
 - global, 13
 - listing, in order, 13
 - maximum length, 22
 - scope, 13
 - using double dots in, 31
- macros
 - calling, 17
 - procdsn, 18, 23, 27, 28, 30
 - procdsns, 31
 - procfile, 18, 25
 - procvar, 19, 20, 22, 27
 - procvars, 27
- making lists, 6, 11
- maximum length of macro variable, 22
- operators
 - concatenation, 7, 22
 - like, 9, 14
- options procedure, 15
- problems, 22
- proc
 - catalog, 18
 - contents, 18
 - freq, 19
 - options, 15
 - summary, 19
 - univariate, 19
- procdsn macro, 18, 23, 27, 28, 30
- procdsns macro, 31
- processing
 - data sets, 11, 18, 23, 28, 30
 - filenames, 25
 - files, 18, 25
 - items in macro array, 30
 - macro variables, 13
 - options, 15
 - variables, 19, 20
- procfile macro, 18, 25
- procvar macro, 19, 20, 22, 27

- procvars macro, 27
- scan function, 28
- scl functions, 25
- scope of macro variables, 13
- select
 - distinct, 11
 - into, 9
 - statement, 8
 - text, sql, 8
 - using calculated columns in, 16
- sql
 - dictionary
 - columns, 20
 - dictionaries, 11
 - macros, 13
 - options, 15
 - tables, 23
 - distinct, 11
 - select, 8
 - using calculated columns in, 16
 - selecting text, 8
 - syntax, 4
- summary
 - list processing, 10
- summary procedure, 19
- syntax
 - cat functions, 7
 - sql, 4
- templates
 - macro, 17
 - module, 31
- testing, 18, 19, 21, 24, 26, 31
- univariate procedure, 19
- upper bound of macro array, 31
- using arrays of macro variables, 30
- using calculated columns in sql, 16
- using macro do loops, 27
- vertical bars, 7
- vocabulary, 3
- workarounds, 22
- writing constant text, 8
- writing macro calls, 17