
with Proc SQL is also simple, yet powerful. This Hands On Workshop

Introduction

This paper reviews the theory of programming, how to process every itemOverview
in a list using the simplicity of Proc SQL select into list processing with
SQL’s dictionary tables. It provides basic answers to these questions:

1. How do I process every column in a dataset?

2. How do I process every file in a folder?

3. How do I process every member in a libref?

or
4. How do I process every item in a list?

Students are expected to have the following minimum background:Prerequisites

programming : three to seven years

data step : data structure allocation
with attribute or length statements

macro language : allocate macro variables
write macros with one or more steps

procedures : Contents, Print

This paper cover the following topics:Topics

• programming theory: vocabulary

• Proc SQL syntax

• list processing (dynamic programming) with dictionaries:

– columns

– dictionaries, v9

– filenames, not an sql dictionary, read with scl functions

– macros

– options, v9: group

– tables

• v9 concatenation functions: catt, catx

Basics and Concept Review

This section provides background information.Overview

We communicate in a natural language English (or Chinese, Dutch, French,Programming Theory
and Vocabulary or German) about the artificial language SAS. These computer science

terms and concepts are used throughout this paper.

Terms Used Here
program data structure (also: metadata)

algorithm
data structure attributes declarative, information

(compile-time) statements
name : variable or column
type : character or numeric
length: in bytes

character: 1–32,767
numeric : 1– 8

format
label
see Online Help: Index:

declarative DATA step statements
organization executable: action or control statements

array: has numbered elements
do I=1 to dim(array-name);
put array-name(I);

list : has unnumbered items
do over array-name;
put array-name;

drop, keep
retain
where

algorithm input: libref, data, variable
process perform actions
output: libref, data, to log or list

There are five SQL statements:SQL syntax

1. proc

2. create

3. describe

4. select

5. quit
ProcSQL-0-syntax.sas

1 PROC SQL;
2 PROC SQL noprint;
3 create table table-name as
4 query-expression
5 <order by order-by-item
6 <, ... order-by-item>>;
7 describe table table-name <, ... table-name>;
8 select <distinct> object-item
9 <function>(object-item)

10 <, ... object-item>
11 into :macro-variable <separated by ’ ’>
12 :macro-variable-A, :macro-variable-B
13 :macro-variable1 - :macro-variable&SysMaxLong.
14 from Libref.Data
15 where ColumnChar eq ’value’
16 and ColumnChar2 eq ’value2’
17 or ColumnNum eq <num-value>
18 group by group-by-item
19 <, ... group-by-item>>
20 having sql-expression
21 order by order-by-item
22 <, ... order-by-item>>;
23 quit;

The keyword select has one required clause, from, and five optional
clauses: into, where, group by, having, and order by, which
may be viewed conceptually in this hierarchy:

select
into
from

where
group by

having
order by

SQL Basic Processes

Proc SQL can be used to do each of the basic processes:Overview

• list data structure

• list data

• list only subset

• create data

• make unique list

Proc SQL produces a data structure listing in the log like Proc ContentsListing Data Structure
does in the listing. Instead of the data = option, SQL has the describe
table statement.

ProcSQL-1-describe-table-libref-data.sas
1 PROC Contents data = SAShelp.Class;
2

3 PROC SQL; describe table SAShelp.Class;
4 quit;

Proc SQL works like Proc Print. In the Print method the object is referredListing Data
to with the data = option. In SQL the object reference is the from clause.
Star — asterisk (*) — means all variables.

ProcSQL-1-select-star.sas
1 PROC Print data = SAShelp.Class;
2 var _all_;
3

4 PROC SQL; select * /* _all_ */
5 from SAShelp.Class;
6 quit;

The where statement is available in all procedures. This shows it as a dataListing Subsets
set option. The SQL select ... from statement has a where clause.

ProcSQL-1-select-subset.sas
1 PROC Print data = SAShelp.Class
2 (where = ( Sex eq ’F’
3 and Age ge 14));
4 var Name Age;
5

6 PROC SQL; select Name, Age
7 from SAShelp.Class
8 where Sex eq ’F’
9 and Age ge 14;

10 quit;

A common task is to copy a permanent data set from a permanent storageCreating Data
library to the work library. The SQL statement create table provides a
similar data manipulation environment.

ProcSQL-1-create-table-libref-data.sas
1 DATA Work.Class;
2 set SAShelp.Class;
3

4 PROC SQL; create table Work.Class as
5 select *
6 from SAShelp.Class;
7 quit;

The SQL select statement has a distinct function, which can be usedMaking Unique List
to collapse many instances of variable values into a unique list.

ProcSQL-1-unique.sas
1 PROC Sort data = SAShelp.Class
2 (keep = Age)
3 nodupkey
4 out = Work.UniqueAge;
5 by Age;
6

7 PROC SQL; create table UniqueAge as
8 select distinct Age
9 from SAShelp.Class

10 quit;

There are several differences between the syntax of proc SQL and otherSummary of Basic
SQL procedures.

The two most important to note are that column (variable) names are sep-
arated by commas, and dictionary tables’ values are upper case.

select : use comma as delimiter between column names

wrong : select Column1 Column2 Column3

right : select Column1, Column2, Column3

where : values in dictionary tables are upper case

wrong : where Libname eq ’SAShelp’

right : where Libname eq ’SASHELP’

These papers provide an introduction to SQL: Hermansen [26, sugi22.035]References
Ronk [34, sugi29.268] Wells [39, sugi26.105] Winn, Jr. [43, sugi22.067]

The v9 concatenation functions are used here to replace the double bangConcatenation
Functions (!!), also known as exclamation point, or double vertical bar (||).

Program : This program compares the concatenation operator and cat* func-
tions.

demo-cat-functions.sas
1 DATA Work.Cat_functions;
2 attrib Verb length = $16
3 Text length = $32
4 Prefix length = $ 8
5 Infix length = $ 8
6 Suffix length = $ 8
7 One length = 4
8 Two length = 4
9 Three2 length = 8;

10 Prefix = ’prefix- ’;
11 Infix = ’ *infix* ’;
12 Suffix = ’ -suffix’;
13 One = 1;
14 Two = 2;
15 Three2 = 3.2;
16 Verb = ’concat’;
17 Text = Prefix !! Infix !! Suffix ;
18 output;
19 Verb = ’concat trim’;
20 Text = trim(Prefix) !! trim(Infix) !! trim(Suffix) ;
21 output;
22 Verb = ’concat left trim’;
23 Text = trim(left(Prefix)) !! trim(left(Infix))
24 !! trim(left(Suffix)) ;
25 output;
26 Verb = ’cat’;
27 Text = cat(Prefix,Infix,Suffix) ;
28 output;
29 Verb = ’catt’;
30 Text = catt(Prefix,Infix,Suffix) ;
31 output;
32 Verb = ’cats’;
33 Text = cats(Prefix,Infix,Suffix) ;
34 output;
35 Verb = ’catx’;
36 Text = catx(’+’,Prefix,Infix,Suffix) ;
37 output;
38 Verb = ’catx numbers’;
39 Text = catx(’+’,One,Two,Three2) ;
40 output;
41 stop;
42 Proc Print;

Listing : Notice the presence of double spaces in obs one and four.
demo-cat-functions.lst

1 Obs Verb Text
2

3 1 concat prefix- *infix* -suffix
4 2 concat trim prefix- *infix* -suffix
5 3 concat left trim prefix-*infix*-suffix
6 4 cat prefix- *infix* -suffix
7 5 catt prefix- *infix* -suffix
8 6 cats prefix-*infix*-suffix
9 7 catx prefix-+*infix*+-suffix

10 8 catx numbers 1+2+3.2

Writing Constant Text

SQL provides the ability to select text as well as columns in its queries. InOverview
this section we examine first a simple example of selecting text and then
see how to place both values from columns and surrounding text which
comprise complete SAS statements into a macro variable, which is used to
submit the text for processing.

In this section:
Dictionary Dictionaries . . . . . . . . . . . . . . . . . . . . . 11
Dictionary Macros . . . . . . . . . . . . . . . . . . . . . . . . 13
Dictionary Options . . . . . . . . . . . . . . . . . . . . . . . 15

The select select statement accepts strings as one of its arguments; eachSelecting Text
string can be either single- or double-quoted, which allows the use of macro
variables.

ProcSQL-2-select-constant-text.sas
1 %Let Rows = rows;
2

3 PROC SQL; select MemName, ’has’, Nobs, "&Rows."
4 from Dictionary.Tables
5 where LibName eq "SASHELP"
6 and MemName eq "CLASS"
7 and MemType eq "DATA";
8 quit;

Note: the length of column MemName is 32, which accounts for the wide
space between the words Class and has.

ProcSQL-2-select-constant-text.sas
1 Number of
2 Physical
3 Member Name Observations
4 ---------------------------------------------------------
5 CLASS has 19 rows

This example shows how to concatenate concatenate text and variableCreating a Macro
Variable value, using the catt (concatenation, trimmed) function and put that text

into a macro variable.

For each row the statement is:

Proc Contents data = SAShelp.MemName;

ProcSQL-2-select-text-into-List.sas
1 %Let List = *missing-;* initialize for no rows selected;
2

3 PROC SQL; select catt(’Proc Contents data = SAShelp.’
4 ,MemName
5 ,’;run;’
6 )
7 into :List separated by ’ ’

8 from Dictionary.Tables
9 where LibName eq "SASHELP"

10 and MemName like "V%"
11 and MemType eq "VIEW";
12 quit;
13 &List.;* execute statements in mvar List;
14

Notes:

into : The into clause creates the macro variable named after the colon.
Notice the the last argument of the catt function is a semicolon and
that the delimiter of separated by is space. This could also be:
catt(...,Memname) ...separated by ’;’.

like : The like operator chooses only names beginning with ’V’; these
tables correspond to sql dictionaries.

quit : The statements in the macro variable are procedure statements, there-
fore they must be executed after the quit statement.

These statements are in the macro variable List.List Processing
Statements ProcSQL-2-select-text-into-List.lst, snip 1

1 Proc Contents data = SAShelp.VALLOPT;
2 Proc Contents data = SAShelp.VCATALG;
3 ...
4 Proc Contents data = SAShelp.VVIEW;

Output from statements in the macro variable List.Program Output
ProcSQL-2-select-text-into-List.lst, snip 2

1 The CONTENTS Procedure
2

3 Data Set Name SASHELP.VALLOPT Observations .
4 Member Type VIEW Variables 6
5 Engine SQLVIEW Indexes 0

SAShelp views associated with SQL dictionaries are reviewed by DiIorioReferences
and Abolafia [15, sugi29.237].

Compare with program ProcSQL-D-Dictionaries-list-describe-table below,
which lists SQL dictionaries.

There are several steps in writing statements to a macro variable and exe-List Processing
Summary cuting them:

1. input: data structure

(a) identify the input table

(b) examine its data structure

2. process: expected text

(a) concatenation of text and values into macro variable

i. clarify the text surrounding the variable value(s):
prefix, infix(es), suffix

ii. identify subset, if any; are values in ALL CAPS?

iii. volume of text in macro calls: can information be gotten by
the subroutine macro rather than passed as a parameter?
this caveat is especially important when passing long char-
acter variables such as data set or variable labels.
see: ¶ problems below

(b) remember closure or delimiter:
clause(s): comma, space, or other
statement(s): semicolon (;)
step boundary: run;

(c) execute the statements: SQL: before quit;
SAS: procedures, macros: after quit;

3. output: consider ODS

In the next sections we use this list processing check list to examine several
of the more commonly used dictionaries.

The first examples in the next section — dictionaries, macros and optionsCommentary
— illustrate writing constant text. In the second section we look at how
to write macro calls to generate more complicated amounts of text when
reading columns (variables), filenames, and tables.

Dictionary Dictionaries

SQL provides a meta-dictionary, Dictionary.Dictionaries, which contains aOverview
description of all other dictionaries.

Can we use this meta-dictionary to write statements which describe all the
other dictionaries?

The task is to write these statements:Problem Statement

describe table Dictionary.MemName;

Dictionary.Dictionaries is unique on MemName and Name (Column Name).

We need to make a data set (list) unique on MemName.

The describe table statement lists the data structure.Review Data Structure
Note: V9 only.

ProcSQL-D-Dictionaries-describe-table.sas
1 PROC SQL; describe table Dictionary.Dictionaries;
2 quit;

The data structure is written to the Log:
ProcSQL-D-Dictionaries-describe-table.log

1 create table DICTIONARY.DICTIONARIES
2 (
3 memname char(32) label=’Member Name’,
4 memlabel char(256) label=’Dataset Label’,
5 name char(32) label=’Column Name’,
6 type char(4) label=’Column Type’,
7 length num label=’Column Length’,

This program makes a data set with the names of all the SQL dictionariesProgram
using select distinct function, writes describe table statements
for each, then executes those statements. The log contains the data struc-
ture of each SQL dictionary.

ProcSQL-D-Dictionaries-list-describe-table.sas
1 *note: need v9 for D.Dictionaries;
2 Proc SQL; create table List as
3 select distinct MemName
4 from Dictionary.Dictionaries;
5 select catt(’describe table Dictionary.’
6 , MemName
7 ,’;’
8 )
9 into :List separated by ’ ’

10 from List ;
11 &List.;
12 quit;
13 run;

The statements in the macro variable are SQL statements, therefore they
must be executed before the quit statement.

These are the statements in the macro variable List:List Processing
Statements ProcSQL-D-Dictionaries-describe-table.lst

1 describe table Dictionary.CATALOGS;
2 describe table Dictionary.CHECK_CONSTRAINTS;
3 ...
4 describe table Dictionary.VIEWS;

SQL table descriptions are written to the log.Program Output
ProcSQL-D-Dictionaries-describe-table.log

1 NOTE: SQL table DICTIONARY.CATALOGS was created like:
2 ...
3 NOTE: SQL table DICTIONARY.CHECK_CONSTRAINTS was created like:
4 ...
5 NOTE: SQL table DICTIONARY.VIEWS was created like:

Compare with program ProcSQL2-select-text-into-List above, which listsCommentary
SAShelp views.

This program illustrates a two-statement solution to this problem. See pro-
gram ProcSQL-D-Options-list-groups below for a single-statement solution.

Dictionary Macros

SQL maintains a list of all macro variables.Overview

One problem with global macro variables is checking their values.

The statement %Put user writes an unsorted list of macro variable names
and values to the log.

Put-User.log
1 1 %Let A = 1;
2 2 %Let b = 22;
3 3 %Let z = end of list;
4 4 %Put _user_;
5 GLOBAL Z end of list
6 GLOBAL A 1
7 GLOBAL B 22

The task is to write a %put statement for each macro variable. Hard-codedProblem Statement
this would be:

%put a: &a.;
%put b: &b.;
%put z: &z.;

These statements list the data structure of Dictionary.Macros:Review Data Structure
ProcSQL-D-Macros-describe-table.sas

1 PROC SQL; describe table Dictionary.Macros;
2 select distinct Scope
3 from Dictionary.Macros;
4 quit;
5 run;

Here is the data structure:
ProcSQL-D-Macros-describe-table.log

1 create table DICTIONARY.MACROS
2 (
3 scope char(32) label=’Macro Scope’,
4 name char(32) label=’Macro Variable Name’,
5 offset num label=’Offset into Macro Variable’,
6 value char(200) label=’Macro Variable Value’

The values of scope are:
ProcSQL-D-Macros-describe-table.lst

1 Macro Scope
2 -----------
3 AUTOMATIC
4 GLOBAL

This program writes the %put statements into the macro variable namedProgram
List and executes them.

ProcSQL-D-Macros-select-into-list-ordered.sas
1 %Let Z = the last one;
2 %Let A = 1st item;
3 %Let M = middle;
4 Proc SQL; select catx(’ ’,’%Put’,Name
5 ,’:’ ,Value
6 ,’;’
7 )
8 into :List separated by ’ ’
9 from Dictionary.Macros

10 where Scope eq ’GLOBAL’
11 and not(Name like ’SQL%’)
12 order by Name;
13 quit;
14 &List.;

Notes:

quit : The statements in the macro variable are SAS statements, therefore
they must be executed after the quit statement.

Exclusion : The clause not(Name like ’SQL%’) excludes the macro variables
created by the proc sql statement: SqlObs, SqLoops, SqlXobs, and
SqlRc.

The statements in macro variable List:List Processing
Statements ProcSQL-D-Macros-select-into-list-ordered.lst

1 %Put A : 1st item ;
2 %Put M : middle ;
3 %Put Z : the last one ;

The statements written to the log are a sorted list of global macro variables.Program Output
ProcSQL-D-Macros-select-into-list-ordered.log

1 14 &List.;
2 A : 1st item
3 M : middle
4 Z : the last one

Dictionary Options

Let’s take a look at Dictionary.Options. These next programs show how toOverview
find out:

• definitions and values of each option

• what options are in each group

The task is to write these statements:Problem Statement

Proc Options define value option = OptName;

What is the data structure?Review Data Structure
ProcSQL-D-Options-describe-table.log

1 create table DICTIONARY.OPTIONS
2 (
3 optname char(32) label=’Option Name’,
4 opttype char(8) label=’Option type’,
5 setting char(1024) label=’Option Setting’,
6 optdesc char(160) label=’Option Description’,
7 level char(8) label=’Option Location’,
8 group char(32) label=’Option Group’

Note: group is available in v9.

This program writes text for each option.Program
ProcSQL-D-Options-define-value.sas

1 PROC SQL; select catt(’Proc Options define value option = ’
2 ,OptName
3 ,’;’
4 )
5 into :List separated by ’ ’
6 from Dictionary.Options;
7 quit;
8 &List.;
9 run;

These statements are in macro variable List:List Processing
Statements ProcSQL-D-Options-define-value.lst

1 Proc Options define value option =APPLETLOC;
2 Proc Options define value option =ARMAGENT;
3 ...
4 Proc Options define value option =XCMD;

Here is the list of values and definitions for the first option written to the log.Program Output
ProcSQL-D-Options-define-value.log

1 Option Value Information For SAS Option APPLETLOC
2 Option Value: C:\Program Files\SAS\Shared Files\applets\9.1
3 Option Scope: SAS Session
4 How option value set: Config File(s)
5 Option Definition Information for SAS Option APPLETLOC
6 Group= ENVFILES
7 Group Description: SAS library and file location information
8 Description: Location of Java applets
9 Type: The option value is of type CHARACTER

In program ProcSQL-D-Dictionaries-list-describe-table above, using theOptions in Each
Group distinct function, we created a table of the unique dictionaries before we

could write the text of the describe table statements. We can reduce
two statements to one by creating a named column with the distinct val-
ues and then refer to the new column in our text concatenation. This trick
requires that we make two macro variables, Item and List, for each of the
columns: Group as Item, and text.

This example shows that the variable value can be used more than once,Using Calculated
Columns and there is no limit to how much text can be concatenated, either before

or after use of the variable value.
ProcSQL-D-Options-list-distinct-groups.sas

1 Proc SQL; select distinct Group as Item,
2 catx(’ ’,’%Put Group: ’
3 , calculated Item
4 ,’;Proc Options group =’
5 , calculated Item
6 ,’;run;’
7 )
8 into :Item, :List separated by ’ ’
9 from Dictionary.Options;

10 quit;
11 &List.;

The calculated keyword indicates that the column Item has been cre-
ated — distinct Group as Item — and is not in the the table being
read: from Dictionary.Options.

Note that the column Item retains the label of Group.List Processing
statements ProcSQL-D-Options-list-distinct-groups.lst

1 Option Group
2 ----------------------------------------------------------
3 COMMUNICATIONS %Put Group: COMMUNICATIONS ;
4 Proc Options group = COMMUNICATIONS ;run;
5 DATAQUALITY %Put Group: DATAQUALITY ;
6 Proc Options group = DATAQUALITY ;run;

The run statement ensures that the the %Put Group: statement is writ-
ten before each group.

This is the first group written to the log.Program Output
ProcSQL-D-Options-list-groups.log

1 12 &List.;
2 Group: COMMUNICATIONS
3 SAS (r) Proprietary Software Release 9.1 TS1M3
4

5 NOAUTOSIGNON SAS/CONNECT remote submit will not
6 automatically attempt to SIGNON
7 COMAMID=TCP Specifies the communication access method
8 to be used for SAS distributed products

Writing Macro Calls

Macros provide the ability to generate more than a single statement andOverview
also conditional processing of parameter values.

In the following sections we examine program templates for processing the
three most commonly used lists:

• columns (variables)

• tables (data set names)

• files

Several macro templates are provided for your later usage that show mini-
mal processing of the parameters passed.

In this section:
Macros . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 18
Dictionary Columns . . . . . . . . . . . . . . . . . . . . . . . 20
Dictionary Tables . . . . . . . . . . . . . . . . . . . . . . . . 23
Filenames . . . . . . . . . . . . . . . . . . . . . . . . . . . . 25

Macros

This macro is an example program that processes a data set in a library.Macro ProcDsn
ProcDsn.sas

1 /* name: ProcDsn.sas
2 description: process a data set
3 purpose : called by SQL list processing
4 usage:
5 %ProcDsn(Data=SAShelp.PrdSal2,MemType=data);
6 ******/
7 %Macro ProcDsn(Data = &Libname..&MemName. /* global */
8 ,MemType = data /* catalog view */
9 ,Testing = 0);

10

11 *upcase for logical expression evaluation;
12 %let MemType= %upcase(&MemType.);
13

14 %let Testing = %eval(&Testing
15 or %sysfunc(getoption(mprint)) eq MPRINT);
16

17 %if &Testing %then %put _local_;
18

19 %if &MemType. eq CATALOG %then %do;
20 Proc Catalog catalog = &Data.;
21 contents;
22 quit;
23 %end;
24 %else %if &MemType. eq DATA %then %do;
25 Proc SQL; describe table &Data.;
26 quit;
27 %end;
28 %else %if &MemType. eq VIEW %then %do;
29 Proc Contents data = &Data.;
30 run;
31 %end;
32 %else %Put &Data. type unknown: &type.;
33

34 title3 "&Data. type: &MemType.";
35 %Mend;

This macro is an example program that processes a file in a folder.Macro ProcFile
ProcFile.sas

1 /* name: ProcFile.sas
2 description: process a file in a folder
3 purpose : called by SQL list processing
4 ******/
5 %Macro ProcFile(Directory = &Folder. /*global*/
6 ,Filename = );
7 %if %index(&Filename.,.) %then
8 %put note: filename.ext:<&FileName.>;
9 %else %put note: other<&FileName.>;

10 run; %mend;
11

This macro is an example program that processes character and numericMacro ProcVar
variables with different procedures.

ProcVar.sas
1 /* name: ProcVar.sas
2 description: process a variable
3 purpose : called by SQL list processing
4 usage:
5 %ProcVar(Data=SAShelp.Class,Name=Sex,Type=char)
6 NOTE: SQL dictionary.tables.type in (char,num)
7 compare Contents.type in (1==num,2==char)
8 ******/
9 %Macro ProcVar(Data = &Libname..&MemName. /* global */

10 ,Name = Sex
11 ,Type = char
12 ,Testing = 0);
13

14 *lowcase for logical expression evaluation;
15 %let Name = %lowcase(&Name.);
16

17 %Let Testing = %eval(&Testing
18 or %sysfunc(getoption(mprint)) eq MPRINT);
19

20 %If &Testing. %then %put _local_;
21

22 %if &Type. eq char %then %do;
23 Proc Freq data = &Data.;
24 tables &Name.;
25 %end;
26 %else %if &Type. eq num %then %do;
27 Proc Summary data = &Data. print;
28 *Proc Univariate data = &Data.;
29 var &Name.;
30 %end;
31

32 title2 "&Data..&Name. type: &Type.";
33 run;%Mend;

Dictionary Columns

Dictionary Columns contains the attributes of each variable in a data set,Overview
including name, type, length, label and format.

The task is to process every variable in a data set, or: every column in aProblem Statement
table.

Review Data Structure
ProcSQL-D-Columns-describe-table.sas

1 PROC SQL; describe table Dictionary.Columns;
2 quit;

ProcSQL-D-Columns-describe-table.log
1 create table DICTIONARY.COLUMNS
2 (
3 libname char(8) label=’Library Name’,
4 memname char(32) label=’Member Name’,
5 memtype char(8) label=’Member Type’,
6 name char(32) label=’Column Name’,
7 type char(4) label=’Column Type’,

This program processes every variable in a data set, using the macro Proc-Program
Var.

ProcSQL-D-Columns-select-into-list.sas
1 options mprint ; *testing: view macro statements;
2 options source2; *testing: echo include to log;
3

4 %Let SQLprint = print;*testing;
5 *Let SQLprint = noprint;
6

7 %Let LibName = SAShelp;
8

9 %Let MemName = Class;
10 *Let MemName = PrdSal2;
11 *Let MemName = PrdSal3;
12 *Let MemName = PrdSale;
13

14 %Let MemType = data;
15 *Let MemType = view;
16

17 %Include ’ProcVar.sas’;
18

19 PROC SQL &SQLprint.;
20 select catt(’%ProcVar(name=’ , Name
21 ,’,type=’ , Type
22 ,’)’
23 )
24 into :List separated by ’ ’
25 from Dictionary.Columns
26 where LibName eq "%upcase(&LibName.)"
27 and MemName eq "%upcase(&MemName.)"
28 and MemType eq "%upcase(&MemType.)";
29 quit;
30 &List.; *execute;

Notes:

Testing : Place an asterisk (*) before the options statements to disable either
or both of the macro option mprint or include option source2.

Change the %Let verb to *Let to disable SQL print of query results.

Parameters : The program has three required parameters:

LibName : The primary parameter is the libref LibName.

MemName : The secondary parameter is the table name, MemName, in the
libref.

MemType : A tertiary parameter, provided for clarity, is the member type,
MemType, in (data, view).

Macro : The %Include statement brings the macro definition into the pro-
gram.

These are the statements in the macro variable List. Spaces have beenList Processing
Statements added for clarity.

ProcSQL-D-Columns-select-into-list.lst
1 %ProcVar(name=Name ,type =char)
2 %ProcVar(name=Sex ,type =char)
3 %ProcVar(name=Age ,type =num )
4 %ProcVar(name=Height,type =num )
5 %ProcVar(name=Weight,type =num )

The macro ProcVar generates steps.Program Output
ProcSQL-D-Columns-select-into-list.lst

1 58 &List.; *execute;
2 MPRINT(PROCVAR): Proc Freq data = SAShelp.Class;
3 MPRINT(PROCVAR): tables Name;
4 ...
5 MPRINT(PROCVAR): Proc Freq data = SAShelp.Class;
6 MPRINT(PROCVAR): tables Sex;
7 ...
8 MPRINT(PROCVAR): Proc Summary data = SAShelp.Class print;
9 MPRINT(PROCVAR): var Age;

212121



There are limits associated with this method.Problems

catt : The catt function returns a value whose maximum length is 200 in
proc sql.

RnD-catt-function-warning.log
1 WARNING: In a call to the CATT function, the buffer allocated
2 for the result was not long enough to contain the
3 concatenation of all the arguments. The correct
4 result would contain 200+ characters, but the actual
5 result may either be truncated to 200 character(s) or
6 be completely blank, depending on the calling
7 environment.

macro variable : The maximum length of a macro variable is 216 − 2 = 65534.

These statements show the error.
RnD-macro-variable-length-error.log

1 2 %Let MaxLen = %eval(2**16);
2 3 %Put MaxLen = &MaxLen.;
3 MaxLen = 65536
4 4 %Let List = %sysfunc(repeat(*,&MaxLen.-2));
5 ERROR: The text expression length (65535)
6 exceeds maximum length (65534). The text expression
7 has been truncated to 65534 characters.

These are some workarounds for consideration:Workarounds

• replace catt function with concatenation operator: double bang
or exclamation points: (!!) or double vertical bars (||)

• reduce macro name to one character

• replace named parameters with positional parameters

• eliminate macro parameters when information can be discov-
ered by the subroutine macro; e.g.: variable name is a foreign
key whose associated information — type, etc. — can be gotten
by subroutine macro ProcVar.

Here is an example program showing the workarounds.

1 %let LibName = SAShelp;
2 %let MemName = Class;
3 *previous: Macro ProcVar(data=,name=,...);
4 %Macro V(Name /* positional parameter */
5 /* global macro vars: */
6 ,Data = &LibName..&Memname.
7 ,Testing = 0);
8 *todo: get information: type, format, etc.;
9 ...

10 PROC SQL &SQLprint.;
11 select ’%V(’ !! trim(Name) !! ’)’
12 ...

Dictionary Tables

This program illustrates the concept of having one program define the pa-Overview
rameters of another.

The task is to process every data set in a library. What are the columns
whose values are to be passed to the processing macro?

This program lists the data structure of Dictionary Tables;Review Data Structure
ProcSQL-D-Tables-describe-table.sas

1 PROC SQL; describe table Dictionary.Tables
2 ; quit;
3 run;

ProcSQL-D-Tables-describe-table.log
1 create table DICTIONARY.TABLES
2 (
3 libname char(8) label=’Library Name’,
4 memname char(32) label=’Member Name’,
5 memtype char(8) label=’Member Type’,
6 nobs num label=’Number of Physical Observations’,
7 ...
8 nvar num label=’Number of Variables’,

This program processes every data set in a library, using the macro ProcDsn.Program
ProcSQL-D-Tables-select-into-list.sas

1 options mprint;*testing: view macro statements;
2 options source2;*testing: echo include to log;
3

4 %Let SQLprint = noprint;*production;
5 %Let SQLprint = print;*testing;
6

7 %Let LibName = SAShelp;%*testing;
8

9 * Libname MyLib ’<folder-name>’;
10 *Let LibName = MyLib; *production;
11

12 %Let MemType = ’data’; *NOTE: MUST BE QUOTED;
13 *Let MemType = ’catalog’ ’data’ ’view’;
14

15 %Include ’ProcDsn.sas’;
16

17 %Let List = *empty: no rows selected;
18

19 PROC SQL &SQLprint.;
20 select catt(’%ProcDsn(data=’,MemName
21 ,’,type =’ ,MemType
22 ,’,nobs =’ ,Nobs
23 ,’,nvar =’ ,Nvar
24 ,’)’
25 )
26 into :List separated by ’ ’
27 from Dictionary.Tables
28 where LibName eq "%upcase(&LibName.)"
29 and MemType in (%upcase(&MemType.));
30 quit;

31 &List.;*execute macro calls;
32 %symdel List;

Testing : Place an asterisk (*) before the options statements to disable either
or both of the macro option mprint or include option source2.

Change the %Let verb to *Let to disable SQL print of query results.

Parameters : These parameters are required.

LibName : The primary parameter is the libref, LibName.

MemType : Libraries contain catalogs, data sets, and views. Remember to
provide the value in single or double quotes.

As in the processing of Dictionary.Columns program above, the statementsList Processing
Statements written include most of the list attributes to be passed to the macro.

ProcSQL-D-Tables-select-into-list.lst
1 %ProcDsn(data=ADOMSG,type =DATA,nobs = 458,nvar =6)
2 %ProcDsn(data=ADSMSG,type =DATA,nobs = 426,nvar =6)
3 %ProcDsn(data=AFMSG ,type =DATA,nobs = 1088,nvar =6)
4 ...
5 %ProcDsn(data=ZTC ,type =DATA,nobs =18161,nvar =6)

Program Output
ProcSQL-D-Tables-select-into-list.lst

1 65 &List.;*execute macro calls;
2 SAShelp.ADOMSG nobs: 458 nvar: 6
3 MPRINT(PROCDSN): Proc SQL;
4 MPRINT(PROCDSN): describe table SAShelp.ADOMSG;
5 NOTE: SQL table SASHELP.ADOMSG was created like:

Compare with program ProcSQL-D-Columns-select-into-list, above.

Filenames

Processing a list of filenames is different from the previous examples asOverview
there is no SQL dictionary of folders and filenames. Reading the list of files
using SQL requires some interesting tricks. This example is a polished
version of a program written by Hamilton [25, sugi31.046]

This program processes every file in a folder, using the macro ProcFile. ItProgram
uses the scl functions: filename, dopen, and dread.

ProcSQL-filenames-select-into-list.sas
1 /* Name: ProcSQL-select-into-list-from-filenames.sas
2

3 Requirements : description : Process Each File in Folder
4 purpose : list processing: read file list
5 process each file
6

7 -------------------------------------------------------------------
8 Contexts : program group: list processing
9 program type: routine

10 SAS type: program with macro
11 uses routines: in-program macro %ProcFile
12

13 -------------------------------------------------------------------
14 Specifications: input : folder: directory-specification
15 process: open folder
16 read filenames
17 SQL writes macro calls of ProcFile
18 execute list: ProcFile of each filename
19 output : user-defined in macro ProcFile
20

21 -------------------------------------------------------------------
22 Usage Examples:
23 %Let Folder = c:\;
24

25 Information : concept author: Jack Hamilton,
26 DIGITS and DATES The SQL Procedure Goes "Loopy"
27 Proceedings of the 31st Annual SAS(R) Users
28 Group International Conference, 2006.
29 http://www2.sas.com/proceedings/sugi31/046-31.pdf
30 polishing: Ronald J. Fehd
31 for PNWSUG-2006 Hands On Workshop
32 NOTE: The execution of this query involves performing one
33 or more Cartesian product joins that can not be optimized.
34 *** .................................... */
35

36 %Let Folder = c:\;
37 *Let Folder = .; *here;
38 *Let Folder = %sysget(sasroot); *directory-spec;
39 *Let Folder = %sysget(sasroot)\core\sasexe; *922 files;
40

41 options mprint ;*testing: echo macro statements;
42 options source2;*testing: echo included statements;
43

44 %Let SQLprint = noprint; *production;
45 %Let SQLprint = print; *testing;
46

47 %Include ’ProcFile.sas’;
48

49 Data Work.Digits;
50 length Digit 4;
51 do Digit = 0 to 9;
52 output;
53 end;
54 stop;
55 run;
56

57 * see: from (&&E&Evalue.);
58 %Let E1 = select ones.digit as FileNmbr from Digits

252525

http://www2.sas.com/proceedings/sugi31/046-31.pdf


59 as ones;
60 %Let E2 = select ones.digit + 10 * tens.digit
61 as FileNmbr
62 from Digits as ones, digits as tens;
63 %Let E3 = select ones.digit + 10 * tens.digit
64 + 100 * hundreds.digit
65 as FileNmbr
66 from Digits as ones, digits as tens,
67 digits as hundreds;
68 %Let E4 = select ones.digit + 10 * tens.digit
69 + 100 * hundreds.digit
70 + 1000 * thousands.digit
71 as FileNmbr
72 from Digits as ones, digits as tens,
73 digits as hundreds,
74 digits as thousands;
75 Proc SQL &SQLprint.;
76 select filename(’DirSpec’, "&Folder."),
77 dopen(’DirSpec’) as Dir_id,
78 dnum(calculated Dir_id)
79 into :FileName_Rc, :Dir_id, :NmbrFiles
80 from Digits where Digit = 0;
81 %Let NmbrFiles = &NmbrFiles.; *trim;
82 %Let Evalue = %length(&NmbrFiles.);
83 select catt(’%ProcFile(filename=’
84 ,dread(&Dir_id., FileNmbr)
85 ,’)’
86 )
87 into :List separated by ’ ’
88 from (&&E&Evalue.)
89 where FileNmbr between 1 and &NmbrFiles;
90 quit;
91 %Let FileName_Rc = dclose(&Dir_id.);
92 &List.;
93 %symdel E1 E2 E3 E4 Evalue
94 Folder List SQLprint
95 FileName_Rc Dir_id NmbrFiles;

Testing : Place an asterisk (*) before the options statements to disable either
or both of the macro option mprint or include option source2.

Change the %Let verb to *Let to disable SQL print of query results.

Parameters :

Folder : The primary parameter is the folder name.

Macro : The %Include statement brings the macro definition into the
program.

Data Set : The Digits data set is used to make a larger data set, FileNmbrs,
containing the file-numbers used by the dread function.

Macro Variables E* : Macro variables E1–E4 are necessary for processing lists of 9,
99, 999 or 9999 files.

Select : Two select statments are required: the first to assign a fileref, open
the folder and read the number of files present; the second statement
writes the macro calls.

Using Macro %Do Loops

This section provides two methods which replace the select ...intoOverview
:List clause which is fragile.

These methods are:

• delimited list: one macro variable contains the list of values and each
value is delimited by a special character

• macro array: a sequentially-numbered set of macro variables

These methods are better than the set of workarounds shown above.

Note: Programs shown here are calling routines for macro ProcDsn. The
zip file provided contains similar programs for macro ProcVar:

• ProcVars

• ProcVars-caller

• ProcVars-Test

• ProcVarsDL

• ProcVarsDL-Test

List processing techniques shown here are from Fehd and Carpenter [22,
sgf2007.113].

In this section:
Using Items in Delimited List . . . . . . . . . . . . . . . . . 28
Using Arrays of Macro Variables . . . . . . . . . . . . . . . 30

Using Items in Delimited List

This method stores all values in one macro variable with each item sepa-Overview
rated by a delimiter; the items are retrieved for processing using the scan
function.

demo-delimited-list.log
1 1 %Let List = ItemA * ItemB * itemC;
2 2 %Let Item1 = %scan(&List.,1,*);
3 3 %Put Item1: &Item1.;
4 Item1: ItemA
5 4 %Let Item3 = %scan(&List.,3,*);
6 5 %Put Item3: &Item3.;
7 Item3: itemC

This program processes every data set in a libref, using the macro ProcDsnProgram
and the method of one macro variable containing a delimited list of all val-
ues.

ProcDsnsDL.sas
1 /* name: ProcDsnsDL.sas
2 description: process all data sets in libref
3 using named macro
4 purpose : list processing
5 note : uses delimited list of values
6 usage:
7 %ProcDsns(libref=SAShelp,macroname=ProcDsn);
8 ******/
9 %Macro ProcDsnsDL(LibName = SAShelp

10 ,MemType = data /* or view */
11 ,MacroName = put ProcDsn
12 ,SQLprint = noprint
13 ,Testing = 0);
14

15 %Let Testing = %eval(&Testing
16 or %sysfunc(getoption(mprint)) eq MPRINT);
17

18 %if &Testing. %then %let SQLprint = print;
19

20 %local Dlm I;
21 %Let Dlm = *;
22

23 PROC SQL &SQLprint.;
24 select MemName, MemType
25 into :List_Name separated by "&Dlm."
26 , :List_Type separated by "&Dlm."
27 from Dictionary.Tables
28 where LibName eq "%upcase(&LibName.)"
29 and MemType eq "%upcase(&MemType.)";
30 quit;
31

32 %if &Testing %then %put _local_;
33

34 %do I = 1 %to &SqlObs.;
35 %Let Item_Name = %scan(&List_Name,&I.,&Dlm.);
36 %Let Item_Type = %scan(&List_Type,&I.,&Dlm.);
37 %if &Testing %then %do;
38 %put note2:Item_Name&I: &Item_Name.;
39 %put note2:Item_Type&I: &Item_Type.;
40 %end;
41 %&MacroName.(libname=&Item_Name.,memtype=&Item_Type.);
42 %end;
43 run;
44 %Mend;

This program provides a unit test of the routine ProcDsnsDL.Program: Testing
ProcDsnsDL-Test.sas

1 *name: ProcDsnsDL-Test;
2 options mprint ; *testing: view macro statements;
3 *either of:
4 *autoexec:;
5 *filename Project ’.’;
6 *options sasautos = (Project SASautos);
7 *
8 *or;
9 %Include ’ProcDsnsDL.sas’ ’ProcDsn.sas’;

10 *unit test;
11 *default = SAShelp;
12 %ProcDsns();
13

14 %ProcDsnsDL(Libname=SAShelp,MacroName = ProcDsn);
15

16 *Libname Library ’<directory-specification>’;
17 *ProcDsnsDL(Libname=Library);

Using Arrays of Macro Variables

This method stores each value in a macro variable; the items are retrievedOverview
for processing using an index variable.

demo-macro-array.log
1 1 %Macro DemoMacroArray();
2 2 %let Item1 = valueA;
3 3 %let Item2 = value.B;
4 4 %let Item3 = value C;
5 5 %let Dim = 3;
6 6 %do I = 1 %to &Dim.;
7 7 %Put Item&I.: &&Item&I.;
8 8 %end;
9 9 %Mend;

10 10 %DemoMacroArray
11 Item1: valueA5
12 Item2: value.B
13 Item3: value C
14

This program processes every data set in a libref, using the macro ProcDsnProgram
and the macro array method of allocating a series of sequentially-numbered
macro variables for each value.

ProcDsns.sas
1 /* name: ProcDsns.sas
2 description: process all data sets in libref
3 using named macro
4 purpose : list processing
5 usage:
6 %ProcDsns(libref=SAShelp,macroname=ProcDsn);
7 ******/
8 %Macro ProcDsns(LibName = SAShelp
9 ,MemType = data /* or view */

10 ,MacroName = put ProcDsn
11 ,SQLprint = noprint
12 ,Testing = 0);
13 %local I;
14 %let I = 0;
15

16 %Let Testing = %eval(&Testing
17 or %sysfunc(getoption(mprint)) eq MPRINT);
18

19 %if &Testing. %then %let SQLprint = print;
20

21 PROC SQL &SQLprint.;
22 select MemName, MemType
23 into :MemName1 - :MemName&SysMaxLong.
24 , :MemType1 - :MemType&SysMaxLong.
25 from Dictionary.Tables
26 where LibName eq "%upcase(&LibName.)"
27 and MemType eq "%upcase(&MemType.)";
28 quit;
29

30 %if &Testing %then %put _local_;
31

32 %do I = 1 %to &SqlObs.;
33 %&MacroName.(data=&Libname..&&MemName&I.,memtype=&&MemType&I.);
34 %end;
35 run;
36 %Mend;

Notes on program ProcDsns:

Testing : Fehd [19, nesug07.cc12] shows the discipline of using options to turn
on self-reporting by macro routines.

SysMaxLong : This macro variable contains the largest integer available to the op-
erating system and is used to ensure the upper bound of the macro
array is not truncated.

Usage here replaces any smaller number — :MemName9999— which
may result in allocating less than the number of macro variables of the
rows in the table; i.e.: no macro variable MemName10000 is created.

Double dots : The prefix of a macro variable reference is an ampersand (&); the
suffix of a reference is a dot (.). The dot suffix is necessary for a
two-level data set name: Work.Data:

&Libname..&MemName.

Item reference : A reference to an item in the macro array contains the array name and
the index. The array name must be preceeded by two ampersands
in order to resolve correctly:

&&MemName.&I. :: &MemName1

This program is a module and provides a template for use of the macroProgram: Module
routine ProcDsns.

ProcDsns-caller.sas
1 *name: ProcDsns-caller;
2 options mprint ; *testing: view macro statements;
3 *either of:
4 *autoexec:;
5 *filename Project ’.’;
6 *options sasautos = (Project SASautos);
7 *
8 *or;
9 %Include ’ProcDsns.sas’ ’ProcDsn.sas’;

10 %ProcDsns(Libname=SAShelp,MacroName = ProcDsn);
11

12 *Libname Library ’<directory-specification>’;
13 *ProcDsns(Libname=Library);

Conclusion

Any data set is a candidate for use by list processing. To produce dynamic
programs follow these simple steps:

• identify the data set (table)

• examine its data structure

• identify the variables (columns) that contain parameter values

• develop a program with example code

• use proc sql to write that code as text or macro call, substituting vari-
able names for values

• sit back and watch the log zoom by

Summary : be aware of limitations:

• maximum length of 200 for cat functions

• maximum length of one macro variable: 216 − 2 = 65534

Recommendation : Use select into macro arrays, they are the easiest to debug, test and
understand.

